


IL PRIMO SETTIMANALE DI SOFTWARE SU CARTA

L. 1.000

PER IL TUO PERSONAL COMPUTER

Una pubblicazione della J.soft editrice



PAPER soft

24

sped. in abb. post. gr. 1/70

Anno I - N° 24 - 7 dicembre 1984



**Super catalog
Didattica
della moltiplicazione**



**Devastator
Questionario**



**Missioni Janaro
Squalo
Un gioco di colore**



Taxi



Il coniglietto

Editrice

J.soft

Con la collaborazione del Gruppo Editoriale Jackson

Guida all'input dei programmi sullo ZX Spectrum

Le "parole" comprese tra parentesi graffe indicano i caratteri grafici predefiniti (G), il tasto (numera seguente la G), la necessità di premere il tasto insieme a CAPS SHIFT (eventuale S precedente la G) e il numero di ripetizioni del tasto richieste (eventuale numero all'inizio della "parola"). I caratteri grafici definiti dal programma sono invece indicati da lettere maiuscole (corrispondenti al tasto da usare) sottolineate.

Quando leggete	Premete	Vedrete
{G1}	1	
{G2}	2	
{G3}	3	
{G4}	4	
{G5}	5	
{G6}	6	
{G7}	7	
{G8}	8	
{SG1}	CAPS SHIFT 1	
{SG2}	CAPS SHIFT 2	
{SG3}	CAPS SHIFT 3	
{SG4}	CAPS SHIFT 4	
{SG5}	CAPS SHIFT 5	
{SG6}	CAPS SHIFT 6	
{SG7}	CAPS SHIFT 7	
{SG8}	CAPS SHIFT 8	

Se non siete già in modo G, entrateci schiacciando contemporaneamente CAPS SHIFT e 9

Se dovete uscire dal modo G, schiacciate 9

Quando leggete	Premete	Vedrete
A	A	Simbolo grafico definito nel programma in uso.
B	B	
C	C	
D	D	
E	E	
F	F	
G	G	
H	H	
I	I	
J	J	
K	K	
L	L	
M	M	
N	N	
O	O	
P	P	
Q	Q	
R	R	
S	S	
T	T	
U	U	

Se non siete già in modo G, entrateci schiacciando contemporaneamente CAPS SHIFT e 9

Se dovete uscire dal modo G, schiacciate 9

Guida per l'input dei programmi versioni VIC 20 e C64

Notate che i listati contengono "parole" racchiuse tra parentesi graffe { }. Tali parole rappresentano caratteri di controllo come mostrato nel sottostante riquadro. Se sono precedute da un numero, questo indica il numero di volte che quel tasto deve essere premuto. Se il simbolo è sottolineato deve essere premuto contemporaneamente a SHIFT mentre se è racchiuso da [< >] deve essere

premutato contemporaneamente al tasto COMMODORE. Inoltre, se tra parentesi si trova un carattere alfabetico "solitario", questo dovrà essere premuto contemporaneamente al tasto CONTROL.

Con questo sistema di codifica, sarà molto più agevole copiare i listati senza faticose e dubbie interpretazioni di caratteri grafici e di controllo del cursore o dei colori.

{CLR}	SHIFT CLR/HOME		{CYN}	CTRL 4		[<7>]	G 7	
{HOME}	CLR/HOME		{PUR}	CTRL 5		[<8>]	G 8	
{SU}	SHIFT ↑ CRSR		{GRN}	CTRL 6		{F1}	F1	
{GIU'}	↓ CRSR		{BLU}	CTRL 7		{F2}	F2	
{SIN}	SHIFT ⇐ CRSR		{YEL}	CTRL 8		{F3}	F3	
{DES}	⇒ CRSR		[<1>]	G 1		{F4}	F4	
{RVS}	CTRL 9		[<2>]	G 2		{F5}	F5	
{OFF}	CTRL 0		[<3>]	G 3		{F6}	F6	
{BLK}	CTRL 1		[<4>]	G 4		{F7}	F7	
{WHT}	CTRL 2		[<5>]	G 5		{F8}	F8	
{RED}	CTRL 3		[<6>]	G 6				



		4	Super Catalog di B. Kersey-trad. e adatt. di M. Cerofolini
		6	Didattica della moltiplicazione di A. Mauffrey trad. e adatt. di M. Cerofolini
Extended Basic	 TI-99/4A	7	Devastator di D. R. Arnold trad. e adatt. di E. Re Garbagnati
	 TI-99/4A	10	Questionario di D. Douglas trad. e adatt. di E. Re Garbagnati
16/48K	 Spectrum	13	Missione Lunare di R. Henderson trad. e adatt. di C. Panzalis
16/48K	 Spectrum	16	Squalo! di M. Hubbuk trad. e adatt. di C. Panzalis
Joystick	 C64	19	Taxi di K. Jones trad. e adatt. di F. Sorcina
Joystick	 VIC-20	25	Il coniglietto di A. Cybanski trad. e adatt. di M. Anticoli e S. Albarelli
16/48K	 Spectrum	27	Un tocco di colore di A. Wallis trad. e adatt. di C. Panzalis

J. soft s.r.l.

**DIREZIONE, REDAZIONE,
AMMINISTRAZIONE**

Via Rosellini, 12
20124 MILANO
Tel. (02) 68.88.228

DIRETTORE RESPONSABILE:
Pietro Dell'Orco

COORDINAMENTO TECNICO:
Riccardo Paolillo

REDAZIONE:

Lucio Bragagnolo
Mauro Cristuilib Grizzi

GRAFICA E IMPAGINAZIONE:
Margherita La Noce
Ivana Rossi
Raffaella Toffolatti

FOTOCOMPOSIZIONE:

d&b Via Vignola, 5
Tel. 02/59.85.08
20133 MILANO

CONTABILITÀ:

Giulia Pedrazzini
Flavia Bonatti

**AUTORIZZAZIONE ALLA
PUBBLICAZIONE:**

Tribunale di Milano n° 200
del 14-04-1984

STAMPA:

Elcograf - Beverate (CO)



Rivista associata
all'Unione stampa
Periodica Italiana

PUBBLICITÀ

Concessionario per l'Italia e l'Estero Reina s.r.l.
Via Washington, 50
20046 MILANO
Tel. (02) 49.88.066 (5 linee R.A.)
Tlx 316213 REINA I
Concessionario esclusivo per la
DIFFUSIONE in Italia e Estero
SODIP - Via Zuretti, 25
20125 MILANO

Spedizione in abbonamento
postale Gruppo II/70
Prezzo della rivista L. 1.000
Numeri arretrati L. 2.000

© TUTTI I DIRITTI DI
RIPRODUZIONE O TRADUZIONE
DEGLI ARTICOLI PUBBLICATI
SONO RISERVATI



Super catalog

Questo programma svolge due funzioni. La prima consiste nell'indicare il numero dei settori liberi sul dischetto e la seconda nel permettere la scelta ed il lancio di un programma con il tocco di un solo tasto.

Dopo avere dato RUN appare sul video il Catalog del dischetto ed ogni programma è preceduto da una lettera dell'alfabeto. A sinistra di questa lettera appare un simbolo rappresentante il tipo di file indicato dalla lettera. I simboli che appaiono sono i seguenti:

| indicante un programma Applesoft
> indicante un programma Integer BASIC

" indicante un file TEXT

* indicante un file binario

A questo punto per mandare in esecuzione il programma desiderato è sufficiente digitare la lettera che contrassegna il programma e quest'ultimo verrà lanciato nel modo opportuno in modo automatico: RUN, BRUN, EXEC ecc. ecc.

Se volete invece sapere il numero dei settori liberi nel dischetto battete il simbolo # e il dato richiesto apparirà nella parte destra dello schermo. Per uscire dal programma battere la lettera 'Z'.

```

5  REM  === SUPER CATALOG ===
10  CA$ = "CATALOG":RU$ = "RUN":EX
    $ = "EXEC":BR$ = "BRUN":NO$ =
    "NOMON"
20  COL = 0

22  REM  (COL=LETTERA-COLONNA. P
    UO' ESSERE 1, 3, 4, OR 7. P
    ONI COL=0 SE NON VUOI LA COL
    ONNA COI COIICI)
25  IF COL < > 1 AND COL < > 3 AND
    COL < > 4 AND COL < > 7 THEN
    COL = 0
110  MEM = ( PEEK (978) - ( PEEK (
    978) > 127) * 256) * 256 + 2
    947:AE = 174 - (157 - PEEK
    (978))
120  NS = 403 + 93 * ( PEEK (MEM +
    2925) = 16)
160  ONERR GOTO 360
210  POKE MEM + 1313,24: POKE MEM
    + 1466,23

260  POKE 768,32: POKE 769,12: POKE
    770,253: POKE 771,201: POKE
    772,141: POKE 773,208: POKE
    774,3: POKE 775,76: POKE 776
    ,44: POKE 777,AE: POKE 778,7
    6: POKE 779,60: POKE 780,AE:
    POKE 781,0
310  POKE MEM + 1463,0: POKE MEM +
    1464,3
360  MAX$ = "W":O$ = CHR$ (4): PRINT
    O$;NO$ "C,I,0"
410  SPEED= 255: TEXT : NORMAL : HOME
    : VTAB 1: PRINT : PRINT O$;C
    A$
460  FOR X = 1 TO 45 STEP 2: IF (
    SCRN( 5,X) < > 11) THEN NEXT
    X

```

```

510  LTR = 64:X = X - 1: FOR V = X
    TO 44 STEP 2: NORMAL : VTAB
    V / 2 + 1: HTAB 1
560  LOCK$ = " ": IF SCRN( 0,V) =
    10 THEN LOCK$ = ":"
610  IF SCRN( 1,V) = 1 THEN FILE
    $ = "]"
660  IF SCRN( 1,V) = 2 THEN FILE
    $ = "*"
710  IF SCRN( 1,V) = 9 THEN FILE
    $ = ">"
760  IF SCRN( 1,V) = 4 THEN FILE
    $ = CHR$ (34)
810  LTR = LTR + 1:TEST$ = N$:N$ =
    CHR$ (LTR)
860  IF SCRN( 1,V) = 0 THEN FILE
    $ = " ":N$ = " ": IF N$ = "
    " AND TEST$ < > " " THEN MA
    X$ = TEST$
910  IF TEST$ = " " AND N$ = " " THEN
    VSEL = V / 2 + 1: GOTO 1060
960  IF NOT COL THEN PRINT "
    ";: POKE 50,63 + 192 * (LOCK
    $ = " "): PRINT FILE$;: NORMAL
    : PRINT " ";: INVERSE : PRINT
    N$;: NORMAL : IF N$ < > " "
    THEN HTAB 38: PRINT "."
961  IF COL THEN HTAB COL: INVERSE
    : PRINT N$;: NORMAL : IF N$ <
    > " " THEN HTAB 38: PRINT
    "."
1010  NEXT : NORMAL
1060  IF VSEL = 0 THEN VSEL = 24
1110  IF NOT COL THEN VTAB VSEL
    : HTAB 1: PRINT SPC( 37): HTAB
    6: INVERSE : PRINT " SELECT
    ";: POKE - 16368,0: GET A$
    : VTAB VSEL: HTAB 15: INVERSE
    : PRINT A$;" ";:TSC = SCRN(

```



```

14,2 * VSEL - 2):8SC = SCRN(
14,2 * VSEL - 1): NORMAL
1111 IF COL THEN VTAB VSEL: HTAB
1: PRINT SPC( 37): HTAB COL
: INVERSE : PRINT SPC( B -
COL);"SCEGLI:";
1120 IF COL THEN POKE - 1636B,
0: GET A$: VTAB VSEL: HTAB 1
5: INVERSE : PRINT A$;" ";;
TSC = SCRN( 14,2 * VSEL - 2
):BSC = SCRN( 14,2 * VSEL -
1): NORMAL
1160 IF ASC (A$) = 13 THEN 360
1170 IF A$ = "2" THEN POKE 216,
0: POKE MEM + 1463,12: POKE
MEM + 1464,253: END
1180 IF A$ = "#" THEN GOSUB 500
0: GOTO 1060
1210 IF A$ > MAX$ OR A$ < "A" THEN
1060
1260 IF NOT COL THEN FOR I = 0
TO 46 STEP 2: IF ( SCRN( 5,
I) = TSC AND SCRN( 5,I + 1)
= BSC) AND ( SCRN( 4,I + 1)
> 9 AND SCRN( 6,I + 1) > 9
) THEN 1360
1261 IF COL THEN FOR I = 0 TO 4
6 STEP 2: IF ( SCRN( COL - 1
,I) = TSC AND SCRN( COL - 1
,I + 1) = BSC) THEN 1360
1310 NEXT
1360 IF NOT COL THEN SCT = SCRN(
3,1): IF SCT = 13 OR SCT = 1
4 THEN T$ = RU$
1361 IF COL THEN SCT = SCRN( 1,
1): IF SCT = 1 OR SCT = 9 THEN
T$ = RU$
1410 IF NOT COL THEN IF SCT =
10 THEN T$ = 8R$
1411 IF COL THEN IF SCT = 2 THEN
T$ = 8R$
1460 IF NOT COL THEN IF SCT =
2 THEN T$ = EX$
1461 IF COL THEN IF SCT = 4 THEN
T$ = EX$
1510 FOR J = X / 2 + 1 TO VSEL -
1: VTAB J: HTAB 1: PRINT "
";: NEXT : VTAB VSEL: HTAB
1: CALL - 958
1560 VTAB I / 2 + 1: HTAB 1: INVERSE
: PRINT " ";: HTAB 2: PRINT
T$;;: NORMAL
1610 V = I: FOR H = 7 TO 37: GOSUB
1960: COLOR= TSC: PLOT H,V: COLOR=
8SC - 12 * (BSC = 12 OR BSC =
- 8 * (8SC = 10 OR 8SC = 11
): PLOT H,V + 1
1660 IF 8SC < > 10 OR TSC < >
0 THEN SP = 0
1710 IF 8SC = 10 AND TSC = 0 THEN
SP = SP + 1
1760 IF SP = 2 THEN 1850
1810 NEXT
1820 POKE MEM + 1313,22: POKE ME
M + 1466,21: POKE MEM + 1463
,12: POKE MEM + 1464,253
1850 POKE 216,0: POKE MEM + 1463
,12: POKE MEM + 1464,253
1860 VTAB 22: PRINT : PRINT D$;T
$
1910 END

```

```

1960 8SC = SCRN( H,V + 1):TSC =
SCRN( H,V)
2010 N = 64: IF BSC = 13 THEN N =
80
2060 IF 8SC = 10 THEN N = 32
2110 IF 8SC = 11 THEN N = 48
2160 FOR X = 0 TO 15: IF SCRN(
H,V) = X THEN T$ = T$ + CHR$
(X + N): RETURN
2210 NEXT : RETURN
5000 RESTORE : FOR I = 896 TO 96
0: READ P: POKE I,P: NEXT
5010 AE = 149 - (157 - PEEK (97B
)): POKE 905,AE: POKE 938,AE
5020 D = PEEK (MEM + 485):S = PEEK
(MEM + 487)
5030 POKE 897,S * 16: POKE 911,S
* 16: POKE 898,D: POKE 912,
D
5040 CALL 919: REM $397
5050 R = PEEK (917) + 256 * PEEK
(918):U = NS - R
5060 VTAB 3: HTAB 35: POKE 32,37
: POKE 33,3: PRINT : PRINT "
-----"
5070 INVERSE : VTAB 5: PRINT "
USAT1 ": PRINT " L18ER1":
VTAB 5: PRINT "":U: VTAB 9:
PRINT "":R: TEXT : NORMAL :
RETURN
5080 DATA 1,96,1,0,17,0,145,3
,0,149,0,0,1,0,0,96,1,0,1,23
9,216,0,0,169,3,160,128,32,2
17,3,162,0,142,149,3,142,150
,3,160,56,185,0,149,162,8,10
,144,8,238,149,3,208,3,238,1
50,3,202,208,242,200,192,196
,144,232,96

```

Guida per l'input dei programmi sul TI 99/4A

A causa del sistema di codifica utilizzato, possono capitare in alcune linee delle improvvise "andate a capo" delle quali, naturalmente, non va tenuto conto in fase di digitazione.

Vi potrà capitare, a volte di trovare nei listati di programmi per TI 99/4a alcuni caratteri sottolineati. La sottolineatura rappresenta una particolare codifica dei tasti di controllo: dovrete quindi premere il tasto indicato dal carattere insieme al tasto CONTROL.

Ad esempio, se trovate A dovrete battere CONTROL+A; se trovate e—d, dovrete battere CONTROL+SPAZIO, e così via.

Didattica della moltiplicazione



Con questo programma potete insegnare e/o esercitare i vostri bambini ad eseguire le moltiplicazioni in modo più divertente. Stabilito l'uso che ne volete fare, dovete quindi scegliere il numero di cifre del moltiplicatore e del moltiplicando. Scegliendo il caso dell'insegnamento l'utente viene guidato nelle varie fasi del cal-

colo, dalle domande che gli pone il computer.

L'utente che voglia esercitarsi, deve essere in grado di svolgere l'operazione da solo, ma nel caso commetta più di 4 errori, il computer dimostra il modo per terminare correttamente la moltiplicazione.

```

10 DIM A%(9),B%(9),S%(18)
20 REM *****
   *
   *      MENU      *
   *
   *****

30 HOME : PRINT "PREFERISCI: "
   : HTAB 5: VTAB 5: PRINT "1-C
   HE IO FACCIA LA MOLTIPLICAZI
   ONE?"
40 HTAB 5: VTAB 10: PRINT "2-VUO
   I FARLA DA SOLO?"
50 VTAB 15: PRINT "DAI IL NUMERO
   DELLA DOMANDA": VTAB 20: GET
   A$: PRINT A$: IND% = VAL (A$
   ): IF ASC (A$) < 49 OR ASC
   (A$) > 50 THEN 50
60 HOME : PRINT "IL MOLTIPLICAND
   O E IL MOLTIPLICATORE SON
   O NUMERI DA 1 A 9 CIFRE.
   A TE LA SCELTA"
70 VTAB 10: PRINT "NUMERO DELLE
   CIFRE DEL MOLTIPLICANDO: ";
   GOSUB 540: LA% = VAL (Z$)
80 VTAB 15: PRINT "NUMERO DELLE
   CIFRE DEL MOLTIPLICATORE: ";
   : GOSUB 540: LB% = VAL (Z$)
90 REM *****
   * SCELTA DEI NUMERI *
   *      E      *
   * SCRITTURA SCHERMO *
   *****

100 FOR I = 0 TO 18: S%(I) = 0: NEXT
   : FAUT = 0: HOME : VS = 3: PH =
   25: FOR I = LA% TO 1 STEP -
   1
110 GOSUB 400: IF CH = 0 AND I =
   I THEN 110
120 A%(I) = CH: GOSUB 420: NEXT I

130 VS = 4: VTAB VS: HTAB 22 - LB
   %: PRINT "X "; PH = 25: FOR
   I = LB% TO 1 STEP - 1
140 GOSUB 400: IF CH = 0 AND I =
   1 THEN 140
150 B%(I) = CH: GOSUB 420: NEXT I
   : PV = 5: GOSUB 440

```

```

160 REM *****
   *      CALCOLO      *
   *
   *****

170 FOR I = LB% TO 1 STEP - 1: P
   H = 25 - LB% + I: PV = PV + I
   : IF B%(I) = 0 THEN PV = PV -
   1: GOTO 260
180 IF I < LB% THEN GOSUB 630
190 FOR J = LA% TO 1 STEP - 1
200 MULT% = A%(J) * B%(I): S%(I +
   J) = S%(I + J) + MULT%: MULT%
   = MULT% + RET%: RET% = MULT%
   / 10: CH = MULT% - RET% * 10
   : VS = PV: EN% = 0: IF IND% =
   1 THEN GOSUB 580: GOTO 220
210 GOSUB 480
220 GOSUB 420: IF RET% = 0 THEN
   250
230 CH = RET%: VS = I: IF J < > 1
   THEN EN% = LB% - I: GOSUB 4
   80: GOSUB 420: PH = PH + I: GOTO
   250
240 VS = PV: GOSUB 480: GOSUB 420

250 NEXT J
260 GOSUB 650: IF IND% = 1 AND I
   < > 1 THEN VTAB 21: HTAB
   1: SPEED = 80: PRINT "DECALO
   DI UNA CIFRA VERSO SINISTRA
   ": FOR L = I TO 100: NEXT
   L: SPEED = 255
270 RET% = 0: FOR K = 24 TO 24 -
   LA% - LB% STEP - 1: VTAB 1:
   HTAB K: PRINT " ": NEXT K: NEXT
   I: IF LB% = 1 THEN 360
280 PV = PV + 1: GOSUB 440
290 REM *****
   *      CALCOLO E      *
   *      SOMMA      *
   *****

300 PV = PV + 1: PH = 25: EN% = 0
310 FOR I = LA% + LB% TO 1 STEP
   - 1: RET% = S%(I) / 10: CH =
   S%(I) - RET% * 10: S%(I - 1) =
   S%(I - 1) + RET%:

```



```

320 VS = PV: IF I < > I OR S%(I)
    < > 0 THEN GOSUB 480: GOSUB
    420
330 NEXT I: VTAB 21: IF IND% = 1
    THEN SPEED= 80: PRINT "HO
    FATTO LA SOMMA DELLE ";LB%:"
    LINEE OTTENUTE": SPEED= 255

340 IF IND% = 2 THEN HTAB 29: VTAB
    15: FLASH: PRINT "*****
    ": HTAB 29: VTAB 16: PRINT "
    * BRAVO *": HTAB 29: VTAB 17
    : PRINT "*****": NORMAL

350 IF IND% = 1 THEN GOSUB 650:
    VTAB 20: HTAB 15: PRINT "A
    TE ADESSO "
360 GOSUB 650: VTAB 23: HTAB 10:
    PRINT "ANCORA UNO? ( S / N
    )";: GET B$: IF B$ = "S" THEN
    30
370 IF B$ < > "N" THEN 360
380 END
390 REM *****
    *
    *      ROUTINES      *
    *
    *****
400 REM ** SCELTA CIFRE **
410 CH = INT ( RND (1) * 10): RETURN

420 REM **
430 PH = PH - 1: HTAB PH + EN%: VTAB
    VS: PRINT CH: RETURN
440 REM **
450 VTAB PV: FOR I = 24 - LA% -
    LB% TO 24: HTAB I: PRINT "-"
    ;: NEXT I: RETURN
460 REM *****
    *      TEST      *
    *****
470 VTAB 1: HTAB PH: PRINT RET%:
    RETURN
480 REM ***** TEST
490 IF IND% = 1 THEN RETURN
500 HTAB PH + EN% - 1: VTAB VS: GOSUB
    540
510 IF VAL (Z$) = CH THEN RETURN

520 FAUT = FAUT + 1: IF FAUT > 3 THEN
    IND% = 1: VTAB 21: SPEED= 40
    : PRINT "HAI FATTO QUATTRO E
    RRORI ": VTAB 23: HTAB 7
    : PRINT "FINISCO IO LA TUA M
    OLTIMPLICAZIONE ";: SPEED= 25
    5: RETURN
530 GOTO 500

540 REM *****
    *      INIZIO E      *
    *      SCRITTURA RESULT.*
    *****

550 GET Z$
560 IF ASC (Z$) < 48 OR ASC (Z
    $) > 57 THEN 550
570 PRINT Z$: RETURN
580 REM *****
    *      SPIEGAZIONI      *
    *****

590 HTAB 29: VTAB 6: PRINT SPC(
    8): HTAB 29: VTAB 9: PRINT SPC(
    10)
600 W = A%(J) * B%(I): HTAB 29: VTAB
    6: PRINT B%(I);"X";A%(J);"=
    ";W: HTAB 29: VTAB 9: PRINT
    W;"+";MULT% - W;"=";MULT%
610 HTAB 29: VTAB 12: PRINT "SCR
    IVO ";CH: HTAB 32: VTAB 15:
    PRINT "E ": HTAB 29: VTAB 1
    8: PRINT "RIPORTO ";RET%
620 GOSUB 650: VTAB 21: PRINT "P
    REMI UN TASTO QUALSIASI": VTAB
    23: HTAB 11: PRINT "PER CONT
    INUARE ": GET A$: RETURN

630 REM **
640 FOR K = 1 TO LB% - 1: HTAB 2
    5 - K: VTAB PV: PRINT ".": NEXT
    K: RETURN
650 REM **
660 FOR H = 1 TO 40: POKE 1615 +
    H,160: POKE 1871 + H,160: NEXT
    : RETURN

```



Devastator

Extended Basic

Il solito gioco di fantascienza, dirà qualcuno; questa volta però si tratta di qualcosa di realmente soddisfacente.

L'azione si svolge sulla luna dove venite inviati per annientare l'attacco di astronavi aliene, che altro non sono che un diversivo per nascondere il vero attacco di un'astronave denominata

"Devastator", dotata di un potentissimo cannone-laser, con l'obiettivo di distruggere il pianeta terra. L'unico modo di evitare che la terra si distrugga sotto i vostri occhi (e questo accadrà sul serio in questo programma in Extended Basic), è distruggere almeno 10 alieni in circa 30 secondi, il tempo che vi è rimasto

prima che il laser nemico sia innescato. Per distruggere le loro astronavi non dovrete far altro che posizionare il vostro mirino sul nemico, aiutandovi con i tasti con le frecce (tenete conto però che le astronavi nemiche sono molto più maneggevoli della vostra). Insomma, questo programma, esteticamente eccellente (avete presente l'ultima sequenza di Guerre Stellari, con i "ribelli" che si infilano

nella trincea sulla Morte Nera cercando di distruggerla contrastati dalle astronavi di Darth Vader? E, sempre nello stesso film, avete presente la dimostrazione di "potenza" dell'Impero, con la distruzione di un pianeta a scopo dimostrativo? Ebbebe la coreografia di questo gioco si ispira a queste due suggestive scene, con ottimi risultati!). Buon divertimento e... "Che la Forza sia con voi!"

```

100 ! *****
110 ! *   DEVASTATOR   *
120 ! *****
130 ! TI-99  EXT.BASIC
140 GOTO 190
150 FOR F=12 TO 14 :: CALL COLOR
(F,2,1):: NEXT F :: RETURN
160 FOR F=10 TO 16 :: CALL SCREE
N(F):: NEXT F :: CALL SCREEN(2):
: RETURN
170 FOR V=1 TO 30 :: CALL SOUND(
D1,F1,V,F2,V):: NEXT V :: RETURN

180 FOR ROW=2 TO 7 :: CALL HCHAR
(ROW,22,32,7):: NEXT ROW :: RETU
RN
190 RANDOMIZE
200 DIM E$(13)
210 CALL CLEAR :: CALL SCREEN(2)

220 GOSUB 570
230 GOSUB 1070 :: CALL CLEAR ::
CALL SCREEN(2)
240 FOR H=2 TO 14 :: CALL COLOR(
H,2,2):: NEXT H
250 FOR J=1 TO 4 :: FOR I=1 TO 1
1 :: CALL HCHAR(I,INT(RND*28)+3,
46):: NEXT I :: NEXT J
260 DISPLAY AT(13,1):".....
a      b"....."
270 DISPLAY AT(14,1):"hhhhhhhhhhh
i      jhhhhhhhhhhh"
280 DISPLAY AT(15,1):"ppppppppppq
h      hrpppppppppp"
290 DISPLAY AT(16,1):"ppppppppqph
a"....."bhprpppppppp"
300 DISPLAY AT(17,1):"....."appi
hhhhhhhhhjppb"....."
310 DISPLAY AT(18,1):"....."a'pqp
pppppppppprp'b"....."
320 DISPLAY AT(19,1):"hhhh'i'qpp
pppppppppppr'"jhbbb"
330 DISPLAY AT(20,1):"hhhih'a'"
....."b'hjhhh"

```

```

340 DISPLAY AT(21,1):"hhihha'"
....."bhjh"
350 DISPLAY AT(22,1):"pqhhihhhhh
hhhhhhhhhhhhhhjhhrp"
360 DISPLAY AT(23,1):"qphihhhhhh
hhhhhhhhhhhhhhjhpr" :: DISPLAY A
T(24,1):"ppihhhhhhhhhhhhhhhhhh
hhjpp"
370 FOR J=1 TO 2 :: FOR I=12 TO
14 :: CALL HCHAR(I,INT(RND*6)+14
,46):: NEXT I :: NEXT J
380 DISPLAY AT(2,24):CHR$(120)&C
HR$(121):: DISPLAY AT(3,23):CHR$
(122)&CHR$(136)&CHR$(137)&CHR$(1
23)
390 DISPLAY AT(4,22):CHR$(124)&C
HR$(125)&CHR$(138)&CHR$(139)&CHR
$(125)&CHR$(126)
400 DISPLAY AT(5,22):CHR$(127)&C
HR$(125)&CHR$(140)&CHR$(141)&CHR
$(125)&CHR$(128)
410 DISPLAY AT(6,23):CHR$(129)&C
HR$(142)&CHR$(143)&CHR$(130)
420 DISPLAY AT(7,24):CHR$(131)&C
HR$(132)
430 CALL COLOR(12,6,1):: CALL CO
LOR(13,6,1):: CALL COLOR(14,3,6)

440 FOR F=2 TO 8 :: CALL COLOR(F
,16,1):: NEXT F
450 CALL SPRITE(#2,108,11,80,80)
460 CALL MAGNIFY(LEVEL):: SPEED=
8 :: TOL=30 :: IF LEVEL=3 THEN T
OL=15
470 CALL SPRITE(#1,100,16,100,11
0)
480 A=9 :: B=10 :: C=11
490 T=A :: A=B :: B=C :: C=T
500 CALL COLOR(A,2,5):: CALL COL
OR(B,2,14):: CALL COLOR(C,2,7)
510 CALL MOTION(#2,INT(RND*40-20
),INT(RND*40-20))
520 CALL JOYST(1,X1,Y1):: CALL M
OTION(#1,-Y1*SPEED,X1*SPEED)

```



```

530 CALL COINC(#1,#2,TOL,G):: IF
  G THEN GOSUB 740
540 W=W+1 :: IF W>200 THEN 810
550 GOTO 490
560 ! DEFINIZIONE CARATTERI
570 A$="" :: B$="010204081020408
0" :: C$="8040201008040201"
580 CALL CHAR(95,B$)
590 FOR I=96 TO 112 STEP 8 :: CA
LL CHAR(I,A$):: CALL CHAR(I+1,B$
)
600 CALL CHAR(I+2,C$):: NEXT I
610 FOR I=0 TO 13 :: READ E$(I):
: CALL CHAR(120+I,E$(I)):: NEXT
I
620 FOR I=0 TO 7 :: READ E$(I)::
  CALL CHAR(I+136,E$(I)):: NEXT I

630 DATA 00000000000000F7F,000000
000000F0FE,01030F1F3F7FFFFF
640 DATA 80C0F0F8FCFEFFFF,000101
0103030303,FFFFFFFFFFFFFFFF
650 DATA 00808080C0C0C0C0,030303
0301010100,C0C0C0C080808000

660 DATA FF7F3F3F1F0F0703,FFFEFC
FCF8F0E0C0,7F0F000000000000
670 DATA FEF0000000000000,080066
7C18666810
680 DATA E0F07F7F7FFFFF,0818F8
F8F0F8F0F0,7F7F7F3D1C0E0201
690 DATA F0F0908800180000,03070F
0F0F070703,F0FFFFFFFCFCF8F0

700 DATA 0303010101010101,E0C0C0
C080808000
710 CALL CHAR(108,"00073FE2E2E2F
FFF667F0C1C0000000000E0FC474747F
FFF66FE3038000000000")
720 CALL CHAR(100,"0000000003040
808FF0808040300000080808080E0908
888FF888890E0808080")
730 RETURN
740 ! DISTRUZIONE ASTRONAVE ALIE
NA
750 CALL DELSPRITE(#2):: CALL MO
TION(#1,0,0)
760 CALL SCREEN(15):: CALL SCREE
N(10)
770 CALL SCREEN(2):: FOR DVOL=1
TO 24 STEP 4 :: CALL SOUND(100,-
7,DVOL):: NEXT DVOL
780 CALL SCREEN(2)
790 D=D+1 :: CALL SPRITE(#2,108,
11,INT(RND*192)+1,INT(RND*256)+1
)
800 RETURN
810 IF D<10 THEN 850
820 GOTO 990

```

```

830 FOR I=30 TO 1 STEP -2 :: CAL
L SOUND(-1000,-5,I):: NEXT I ::
RETURN
840 ! DISTRUZIONE TERRA
850 GOSUB 830 :: FOR I=8 TO 0 ST
EP -1 :: CALL HCHAR(7+I,25-I,95)
:: CALL COLOR(8,INT(RND*8)+9,1):
: NEXT I
860 FOR J=1 TO 40 :: NEXT J
870 FOR I=8 TO 0 STEP -1 :: CALL
HCHAR(7+I,25-I,32):: NEXT I
880 GOSUB 160 :: D1=-100 :: F1=-
6 :: F2=110 :: GOSUB 170 :: GOSU
B 160 :: GOSUB 150 :: GOSUB 180

890 J=0 :: I=0
900 DISPLAY AT(1,23+I):CHR$(133)
:: DISPLAY AT(1,26+J):CHR$(133)

910 DISPLAY AT(2,22+I):CHR$(133)
&CHR$(133):: DISPLAY AT(2,26+J):
CHR$(133)&CHR$(133)
920 DISPLAY AT(3,21+I):CHR$(133)
&CHR$(133)&CHR$(133):: DISPLAY A
T(4,25+J):CHR$(133)&CHR$(133)&CH
R$(133)
930 DISPLAY AT(5,22+I):CHR$(133)
&CHR$(133):: DISPLAY AT(5,25+J):
CHR$(133)&CHR$(133)&CHR$(133)::
GOSUB 160
940 DISPLAY AT(6,25+J):CHR$(133)
:: DISPLAY AT(7,23+I):CHR$(133):
: DISPLAY AT(7,27+J):CHR$(133)
950 DISPLAY AT(8,22+I):CHR$(133)
&CHR$(133):: DISPLAY AT(9+J,24):
CHR$(133)&CHR$(133)

960 CALL COLOR(13,9,1):: GOSUB 1
60 :: D1=30 :: F1=-6 :: F2=110 :
: GOSUB 170 :: IF J=1 THEN 980
970 I=-1 :: J=1 :: GOSUB 150 ::
GOSUB 160 :: GOTO 900
980 FOR F=1 TO 100 :: NEXT F

990 CALL DELSPRITE(ALL):: W=0
1000 CALL CLEAR :: CALL SCREEN(2
):: DISPLAY AT(8,2):"NAVI ALIENE
DISTRUTTE: ";D
1010 IF D>HD THEN HD=D
1020 DISPLAY AT(13,2):"MIGLIOR P
UNTEGGIO: ";HD
1030 D=0 :: DISPLAY AT(17,1):"RI
TENTI,CAPITANO? (S/N)"
1040 CALL KEY(0,KEY,ST):: IF ST=
0 THEN 1040
1050 IF (KEY=83)+(KEY=115) THEN C
ALL CLEAR :: GOTO 240
1060 DISPLAY AT(21,8):"A PRESTO.
.." :: FOR I=1 TO 500 :: NEXT I
:: STOP

```

```

1070 FOR J=2 TO 8 :: CALL COLOR(
J,1,1):: NEXT J
1080 PRINT "      D E V A S T A T
O R" :: PRINT :: PRINT
1090 PRINT "LA TUA MISSIONE E'DI
PROTEG-":"GERE LA TERRA DALL'AP
PROSSI-":"MARSII DEI DEVASTATORI.
.."
1100 PRINT : : " ABBATTI ALMENO
10 ASTRONA-":"VI ALIENE AFFINCHE
' I TUOI":"COMPAGNI POSSANO DIST
RUGGERE":"IL D E V A S T A T O R
E ."
1110 PRINT : " RICORDA PERO'CHE
POSSIEDI":"UN TEMPO LIMITATO ENT
RO CUI"
1120 PRINT "COMPLETARE LA TUA MI
SSIONE.": : " POSIZIONA IL MIRINO
CON I"
1130 PRINT "TASTI CON LE FRECCHE.
"
1140 FOR J=2 TO 8 :: CALL COLOR(
J,15,1):: NEXT J
1150 PRINT : : "SCEGLI IL TUO LIV
ELLO DI ":"GIOCO(1,2),CAPITANO?"

```

```

:: ACCEPT AT(23,24)BEEP VALIDAT
E("12")SIZE(1):LEVEL$
1160 LEVEL=5-VAL(LEVEL$)
1170 GOSUB 830
1180 PRINT : : : : : : : : : :
: : : : : : : "IL DEVASTATORE SI
AVVICINA."
1190 PRINT : " ADOCCHIA I TASTI
CON LE ":"FRECCHE E .... ":"P
REPARATI ALLA BATTAGLIA!"
1200 FOR I=1 TO 750 :: NEXT I
1210 RETURN
20000 SUB JOYST(N,X1,Y1):: CALL
KEY(0,K,S):: IF S=0 THEN X1,Y1=0
:: SUBEXIT
20010 IF K=83 THEN X1=-4 :: Y1=0
:: SUBEXIT
20020 IF K=68 THEN X1=4 :: Y1=0
:: SUBEXIT
20030 IF K=69 THEN X1=0 :: Y1=4
:: SUBEXIT
20040 IF K=88 THEN X1=0 :: Y1=-4
20050 SUBEND

```

Questionario



L'anno scorso è scoppiata la mania dei "test": dalle riviste per bambini, ai settimanali di enigmistica, alle trasmissioni televisive, periodicamente compariva il solito Test, sugli argomenti più svariati: da quello psicologico, a quello culinario, da quello politico a quello erotico. Questo originale programma, congegnato per girare sia in Basic che in Extended, permette a chiunque, senza possedere la mente di un

prof. Spaltro, di progettare e stampare questionari semplificati con varie risposte tra le quali è possibile sceglierne solo una. È il classico questionario dal metodo delle crocette sulle caselle, e permette anche di lasciare in bianco la risposta (premendo la barra spaziatrice). Il computer vi guiderà attraverso le varie operazioni e vi garantisco che il risultato finale sarà veramente soddisfacente. Buon lavoro!

```

100 REM *****
110 REM *   QUESTIONARIO   *
120 REM *****
130 REM TI-99/4a      BASIC
140 REM
150 REM

```

```

160 REM
170 OPTION BASE 1
180 DIM N(20),Q$(20),A$(160),R$(
70),R(20,20)
190 DEF U=INT(190/(Q+1))
200 DEF V=A+J
210 CALL CLEAR
220 PRINT "***QUESTIONARIO-SONDA
GGI***": :

```



```

230 PRINT "1 CREAZIONE QUESTIONA
RIO": : "2 INSERIMENTO RISPOSTE":
: "3 STAMPA RESOCONTO": : "4 FINE
LAVORO": :
240 INPUT "INSERISCI LA SCELTA (
1-4)": : Z
250 IF Z>4 THEN 240
260 CALL CLEAR
270 ON Z GOTO 290,750,1300,2880

290 PRINT "SE USI VIRGOLE O ACCE
NTI, CHIUDI LE PAROLE TRA APIC
I.": :
300 INPUT "TITOLO": : T$
310 INPUT "NO. DOMANDE": : Q
320 A=0
330 FOR I=1 TO Q
340 PRINT : "INSERIRE LA DOMANDA
#"; I
350 INPUT "": Q$(I)
360 INPUT "INSERIRE NO. RISPOSTE
": : N(I)
370 PRINT "INSERISCI LA RISPOSTA
": "
380 FOR J=1 TO N(I)
390 INPUT A$(V)
400 IF LEN(A$(V))<64 THEN 430
410 PRINT : "RIGA TROPPO LUNGA:RI
FORMULA.": :
420 GOTO 390
430 NEXT J
440 CALL CLEAR
450 PRINT Q$(I)
460 FOR J=1 TO N(I)
470 PRINT "#"; J; A$(V)
480 NEXT J
490 INPUT "1=CAMBIO DOMANDA, 2=C
AMBIO RISPOSTA, 3=CONFERMA": : Z

500 IF Z>3 THEN 490
510 ON Z GOTO 580,520,610
520 INPUT "VARIAZIONE RISPOSTA #
": : J
530 IF J>N(I) THEN 520
540 INPUT "NUOVA FORMULAZIONE: "
: A$(V)
550 IF LEN(A$(V))<64 THEN 440
560 PRINT : "RIGA TROPPO LUNGA:RI
FORMULA.": :
570 GOTO 540
580 PRINT : "RIFORMULA LA DOMANDA
": "
590 INPUT "": Q$(I)
600 GOTO 440
610 A=A+N(I)
620 NEXT I
630 INPUT "INSERIRE FILENAME PER
ARCHI-VIAZIONE QUESTIONARIO": : Z
$

```

```

640 OPEN #1:Z$,INTERNAL,OUTPUT,F
IXED 192
650 PRINT #1:T$,Q,A
660 FOR I=1 TO Q
670 PRINT #1:Q$(I),N(I)
680 NEXT I
690 FOR I=1 TO A STEP 3
700 PRINT #1:A$(I),A$(I+1),A$(I+
2)
710 NEXT I
720 CLOSE #1
730 GOTO 210
750 INPUT "INSERIRE FILENAME-QUE
STIONA-RIO(SE GIA'IN MEMORIA PRE
ME-RE ENTER SOLTANTO)": : F$
760 IF F$="" THEN 790
770 GOSUB 2620
780 CLOSE #1
790 INPUT "INSERIRE FILENAME PER
ARCHI-VIAZIONE RISPOSTE": : F$
800 OPEN #4:F$,INTERNAL,OUTPUT,F
IXED 192
810 CALL CLEAR
820 PRINT T$: :
830 FOR J=1 TO U
840 PRINT "INSERIRE RISPOSTE:"
850 FOR I=1 TO Q
860 PRINT : Q$(I);
870 INPUT "": Z
880 IF Z=999 THEN 930
890 IF Z=111 THEN 970
900 IF Z<=N(I) THEN 980
910 GOSUB 2820
920 GOTO 860
930 FOR I=1 TO Q
940 P$=P$&"@"
950 NEXT J
960 GOTO 1000
970 Z=-1
980 P$=P$&CHR$(Z+65)
990 NEXT I
1000 INPUT "CONFERMA RISPOSTE (S
/N)?": : Z$
1010 IF (Z$="")+(Z$="S") THEN 114
0
1020 IF Z$<>"N" THEN 1000
1030 INPUT "CAMBIO RISPOSTA IN Q
UEST. #": : Z
1040 IF Z>Q THEN 1030
1050 PRINT : Q$(Z);
1060 INPUT "": Y
1070 IF Y=111 THEN 1110
1080 IF Y<=N(Z) THEN 1120
1090 GOSUB 2820
1100 GOTO 1050
1110 Y=-1
1120 P$=SEG$(F$,1,Z-1)&CHR$(Y+65'
)&SEG$(P$,Z+1,Q)

```

```

1130 GOTO 1000
1140 PRINT #4:P$,
1150 P$=""
1160 INPUT "ALTRI QUESTIONARI (S
/N)?":Z$
1170 IF Z$="N" THEN 1230
1180 IF (Z$<>"")*(Z$<>"S") THEN 1
160
1190 NEXT J
1200 PRINT #4:""
1210 GOTO 830
1220 J=0
1230 IF J=U THEN 1220
1240 FOR J=J+1 TO U
1250 PRINT #4:"9",
1260 NEXT J
1270 CLOSE #4
1280 GOTO 210
1300 INPUT "INSERIRE FILENAME QU
ESTIONA-RIO( SE GIA' IN MEMORIA PR
EME-RE ENTER SOLTANTO):":F$
1310 IF F$="" THEN 1370
1320 GOSUB 2620
1330 FOR I=1 TO A STEP 3
1340 INPUT #1:A$(I),A$(I+1),A$(I
+2)
1350 NEXT I
1360 CLOSE #1
1370 INPUT "INSERIRE FILENAME RI
SPOSTE:":F$
1380 OPEN #2:F$,INTERNAL,INPUT ,
FIXED 192
1390 GOSUB 2690
1400 FOR J=1 TO U
1410 IF R$(J)="9" THEN 1480
1420 FOR I=1 TO Q
1430 X=ASC(SEG$(R$(J),I,1))-63
1440 R(I,X)=R(I,X)+1
1450 NEXT I
1460 NEXT J
1470 GOTO 1390
1480 CLOSE #2
1490 GOSUB 2860
1500 L=66
1510 P=0
1520 A=0
1530 FOR I=1 TO Q
1540 IF L+N(I)+6<65 THEN 1570
1550 GOSUB 2450
1560 GOSUB 2480
1570 S=0
1580 FOR J=3 TO N(I)+2
1590 S=S+R(I,J)
1600 NEXT J
1610 T=S+R(I,2)
1620 PRINT #7: :Q$(I): :TAB(58);
"RISPOSTA INEVASA";
1630 FOR J=0 TO N(I)

```

```

1640 IF J=0 THEN 1660
1650 PRINT #7:TAB(75-LEN(A$(V)))
;A$(V);
1660 Z=R(I,J+2)
1670 Y=84
1680 GOSUB 2560
1690 X=T
1700 Y=101
1710 GOSUB 2540
1720 IF J=0 THEN 1760
1730 X=S
1740 Y=124
1750 GOSUB 2540
1760 PRINT #7
1770 NEXT J
1780 PRINT #7: :TAB(60);"TOTALE
RISPOSTE";
1790 Y=84
1800 Z=T
1810 GOSUB 2560
1820 PRINT #7: :TAB(52);"TOTALE
ESCL. INEVASIONI";
1830 Z=S
1840 GOSUB 2560
1850 PRINT #7: :
1860 L=L+N(I)+8
1870 A=A+N(I)
1880 NEXT I
1890 CLOSE #7
1900 GOSUB 2750
1910 INPUT "RIASSUMI 2 DOMANDE (
S/N)?":Z$
1920 IF Z$="N" THEN 210
1930 IF Z$<>"S" THEN 1910
1940 INPUT "QUALI DOMANDE (ES. 5
,9): ":C,B
1950 IF (C>Q)+(B>Q)+(C>=B) THEN 1
940
1960 OPEN #2:F$,INTERNAL,INPUT ,
FIXED 192
1970 GOSUB 2690
1980 FOR J=1 TO U
1990 IF R$(J)="9" THEN 2050
2000 Y=ASC(SEG$(R$(J),C,1))-63
2010 X=ASC(SEG$(R$(J),B,1))-63
2020 R(X,Y)=R(X,Y)+1
2030 NEXT J
2040 GOTO 1970
2050 CLOSE #2
2060 GOSUB 2860
2070 GOSUB 2450
2080 PRINT #7:"CONDENSAZIONE PER
DOMANDE #:" :TAB(5);Q$(C):TAB(5)
;Q$(B): :
2090 FOR I=1 TO N(C)+1
2100 PRINT #7:TAB(6*I);CHR$(64+I
);
2110 NEXT I

```



```

2120 PRINT #7
2130 FOR I=2 TO N(B)+2
2140 Z=I-1
2150 Y=3
2160 GOSUB 2560
2170 FOR J=2 TO N(C)+2
2180 Y=6*(J-1)+2
2190 Z=R(I,J)
2200 GOSUB 2560
2210 NEXT J
2220 PRINT #7
2230 NEXT I
2240 GOSUB 2750
2250 A=0
2260 FOR I=1 TO C-1
2270 A=A+N(I)
2280 NEXT I
2290 PRINT #7: "TASTI:" : QS(C)
: " A = DOMANDA INEVASA"
2300 FOR J=1 TO N(C)
2310 PRINT #7:TAB(5);CHR$(65+J);
" = ";AS(V)
2320 NEXT J
2330 FOR I=C TO B-1
2340 A=A+N(I)
2350 NEXT I
2360 PRINT #7: QS(B): " 1 = D
OMANDA INEVASA"
2370 FOR J=1 TO N(B)
2380 Y=6
2390 Z=J+1
2400 GOSUB 2560
2410 PRINT #7: " = ";AS(V)
2420 NEXT J
2430 GOTO 1890
2450 P=P+1
2460 PRINT #7:CHR$(12):TAB((132-
LEN(T$))/2);T$;TAB(125);"PAG.";P
: :
2470 RETURN
2480 PRINT #7:TAB(75);"NO. RISPO
STE";TAB(95);"% SUL TOTALE";TAB(
115);"% ESCLUSE INEVASIONI"

```

```

2490 PRINT #7:TAB(75);"-----
----";TAB(95);"-----";TAB(1
15);"-----";
2500 L=4
2510 RETURN
2530 X=1
2540 IF X=0 THEN 2530
2550 Z=INT(R(I,J+2)*1000/X+.5)/1
0
2560 IF POS(STR$(Z),".",1) THEN 2
590
2570 PRINT #7:TAB(Y-LEN(STR$(Z))
);STR$(Z);
2580 RETURN
2590 PRINT #7:TAB(Y+1-POS(STR$(Z
),".",1));STR$(Z);
2600 RETURN
2620 OPEN #1:F$,INTERNAL,INPUT ,
FIXED 192
2630 INPUT #1:T$,Q,A
2640 FOR I=1 TO Q
2650 INPUT #1:QS(I),N(I)
2660 NEXT I
2670 RETURN
2690 FOR J=1 TO U
2700 INPUT #2:RS(J),
2710 NEXT J
2720 INPUT #2:Z$
2730 RETURN
2750 FOR I=1 TO 20
2760 FOR J=1 TO 20
2770 R(I,J)=0
2780 NEXT J
2790 NEXT I
2800 RETURN
2820 CALL SOUND(200,110,0)
2830 PRINT : "FORMULAZIONE ERRATA
.RITENTA.": :
2840 RETURN
2860 OPEN #7:"RS232.BA=1200",VAR
IABLE 132
2870 RETURN
2880 END

```

Missione lunare



16/48K

Benvenuto sulla nostra stazione orbitante, ca-
detto spaziale! Come ben sai, la Terra inizia
a risentire ormai del pesante sfruttamento cui

l'uomo l'ha sottoposta per far fronte alle ne-
cessità della popolazione e del progresso. Si ren-
de dunque necessario ricercare nuove fonti di

approvvigionamento di materie prime, minerali ecc. A motivo di ciò sei stato incaricato di approntare una base lunare, che serva da punto di partenza per successive spedizioni rivolte alla ricerca delle materie richieste. Gli ostacoli che dovrai affrontare sono di tre tipi diversi: la fascia di asteroidi che circonda il satellite; il Mare della Tranquillità, ove è stata inviata una piattaforma galleggiante sulla quale dovrai atterrare; infine le irregolarità del suolo lunare, come è noto caratterizzato da vastissimi

crateri. Nella prima e nella seconda fase (asteroidi e piattaforma) dovrai dunque spostare il tuo modulo lunare con i tasti del cursore per evitare gli scontri e per centrare l'obiettivo. Nella terza fase, sbarcato ormai sulla Luna, dovrai avviarti a piedi alla ricerca del posto adatto. Potrai però salvarti e realizzare il tuo scopo solo se salterai con decisione i crateri, premendo qualsiasi tasto. Ricordati che l'umanità intera ti sta guardando...

```

1 GO SUB 9000
3 LET z$="MISSIONE LUNA"
5 PAPER 0: CLS : PRINT AT 1,9
  ; INK 7;z$; PAPER 5; INK 0;
  FLASH 1; FLASH 0;AT 10,0;"
  Istruzioni per la mission
  e ? ";AT 13,10;"PREMI S /
  N": BEEP .1,10: BEEP .1,5:
  BEEP .1,0: BEEP .5,10
7 PAUSE 0: IF INKEY$="s"
  THEN GO SUB 9500
10 BORDER 3: INK 7: PAPER 0:
  BRIGHT 1: CLS
30 LET xx=INT (RND*25)+5:
  LET yy=0
32 FOR z=1 TO 150
33 IF z>129 THEN PRINT AT 18,
  0;"
      ": FOR n=1 TO 10:
      PRINT ' ': POKE 23692,-1
      : GO TO 50
35 LET a=INT (RND*30)+1
40 PRINT AT 18,a; INK 6;"CD";
  AT 19,a; INK 6;"FE"
43 POKE 23692,-1
48 PRINT ' '
50 PRINT AT yy,xx;"A"
51 BEEP .01,(z/10)-5
52 IF ATTR (yy+1,xx)=70 THEN
  GO TO 9998
60 IF INKEY$="5" AND xx>1
  THEN LET xx=xx-1
70 IF INKEY$="8" AND xx<30
  THEN LET xx=xx+1
100 NEXT z
2000 CLS : FOR n=0 TO 21:
  PRINT AT n,xx;"A";AT n-1,x
  x;" ": BEEP .05,n: BEEP .05
  ,n-10: NEXT n: CLS :
  PRINT AT 7,8;"CONGRATULAZI
  ONI";AT 12,4; PAPER 2;
  INK 7; FLASH 1;"Ora cerca
  di ammarare!": FOR n=1 TO 2
  50: NEXT n

```

```

2030 BORDER 3: PAPER 5: INK 1:
  BRIGHT 1: CLS
2050 LET yy=0: LET xx=INT (RND*1
  0)+20
2060 PRINT AT 20,0;"BBBBBBBBBBBBB
  BBBBBBBBBBBBBBBBBBBBBB";AT 21
  ,0;"{32SG8}"
2070 LET z=0
2100 PRINT AT yy,xx; INK 0;"A":
  BEEP .01,-yy+10
2110 IF INKEY$="5" AND xx>0
  THEN LET xx=xx-1: PRINT
  AT yy,xx+1;" "
2120 IF INKEY$="8" AND xx<31
  THEN LET xx=xx+1: PRINT
  AT yy,xx-1;" "
2130 PRINT AT 19,z; INK 0;"
  {SG8}";AT 20,z;"{SG8}";AT 1
  9,z-1;" ";AT 20,z-1;"B"
2140 LET z=z+1: IF z>30 THEN
  LET z=1: PRINT AT 19,30;"
  ";AT 20,30;"B"
2150 IF yy>18 THEN GO TO 9998
2160 IF NOT yy=18 THEN GO TO 21
  70
2165 IF xx=z OR xx=z-1 OR xx=z+1
  THEN GO TO 3000
2225 PRINT AT yy,xx;" ": LET yy=
  yy+.5
2300 GO TO 2100
3000 FOR n=z TO 31: PRINT AT 18,
  n; INK 0;"A";AT 18,n-1;" ";
  AT 19,n; INK 0;"{SG8}";AT 1
  9,n-1;" ";AT 20,n; INK 0;"
  {SG8}";AT 20,n-1;"B":
  BEEP .03,n: BEEP .03,n-1:
  NEXT n
3005 PAPER 5: INK 4: BRIGHT 1:
  CLS : PRINT AT 5,8;
  PAPER 1; INK 7;"CONGRATULA
  ZIONI";AT 10,3; PAPER 2;
  INK 7; FLASH 1;"Ora tenta
  con i crateri!": FOR n=1
  TO 250: NEXT n

```



```

3010 RESTORE 3300
3015 LET a=20: LET p=0: LET xz=0

3020 LET a$="J      J      J
      J      J      "
3025 LET b$="{SG8}G  H{SG8}G
      H{SG8}G  H{SG8}GH
      {SG8}G  H"
3030 LET c$="{7SG8}G  H{5SG8}G
      H{10SG8}"
3040 LET d$="{8SG8}G  H{8SG8}GH
      {11SG8}"
3050 FOR n=8 TO 21: PRINT AT n,0
      ;"{32SG8}"
3060 NEXT n
3080 PRINT AT 21,0; PAPER 0;"
      {32SG8}"
3090 LET f$="{7SG8}G  H{2SG8}G H
      {SG8}G  H{2SG8}G  H{5SG8}G
      H{6SG8}G  H{2SG8}G  H
      {5SG8}G  H{SG8}G  H
      {4SG8}G  H{4SG8}G  H
      {SG8}G  H{SG8}G  H{11SG8}Ecco
      un bel posto per una Base
      Lunare!{32SG8}"
3100 PRINT AT a,2; PAPER 4;
      INK 0;"I"
3105 IF INKEY$<>" " THEN LET p
      =1
3106 IF xz=135 THEN GO TO 7000
3107 LET xz=xz+1: LET e$=f$(xz
      TO 31+xz)
3110 PRINT AT 4,0;a$
3120 PRINT AT 5,0;b$
3130 PRINT AT 6,0;c$
3140 PRINT AT 7,0;d$
3145 PRINT AT 21,0; PAPER 4;
      INK 0;e$
3150 LET a$=a$(1 TO 32)+a$(1)
3160 LET b$=b$(1 TO 32)+b$(1)
3170 LET c$=c$(1 TO 32)+c$(1)
3180 LET d$=d$(1 TO 32)+d$(1)
3185 LET e$=e$(1 TO 32)+e$(1)
3190 LET a$=a$(2 TO )
3200 LET b$=b$(2 TO )
3210 LET c$=c$(2 TO )
3220 LET d$=d$(2 TO )
3225 LET e$=e$(2 TO )
3230 BEEP .01,-a+20
3235 IF a<>20 THEN GO TO 3245

3240 IF SCREEN$ (21,2)=" "
      THEN PRINT AT 20,1;
      PAPER 4;" ";AT 21,2;
      PAPER 2; INK 7; FLASH 1;"I
      ";AT 0,0; PAPER 6; INK 0;"N
      on sei riuscito a comple
      tare la tua missione

!!!!!! ": BEEP .1,20:
FOR f=1 TO 200: NEXT f:
GO TO 9999
3245 PRINT AT a,2; PAPER 4;
      INK 1;" "
3250 IF p=1 THEN READ a
3260 IF a=0 THEN LET p=0: LET a
      =20: RESTORE
3300 DATA 19,18,17,16,17,18,19,0

3500 GO TO 3100
4000 STOP
7000 REM congratulazioni
7010 PAPER 6: INK 0: BORDER 2:
      CLS
7020 FOR n=1 TO 21: PRINT AT n,0
      ;"A": PRINT AT n-1,0;"
      {SG8}": NEXT n
7030 FOR n=0 TO 30: PRINT AT 21,
      n;"{SG8}A": NEXT n
7040 FOR n=20 TO 0 STEP -1:
      PRINT AT n,31;"A";AT n+1,3
      1;"{SG8}": NEXT n
7050 FOR n=31 TO 0 STEP -1:
      PRINT AT 0,n;"A{SG8}":
      NEXT n
7060 PRINT AT 3,3;"BEN FATTO!!!
      Sei riuscito";AT 6,3;"ad ev
      itare gli ostacoli";AT 9,3;
      "e ad atterrare felicemente
      "
7070 FOR n=1 TO 30: BEEP .01,n:
      BEEP .01,n+1: BEEP .01,n+2
      : BEEP .01,n+3: BEEP .01,n+
      4: NEXT n
7080 PRINT AT 15,0; PAPER 2;
      INK 7; FLASH 1;"Ora la bas
      e puo' esser edificata"
7090 PAUSE 50
7100 PRINT AT 21,0; INVERSE 1;;
      BRIGHT 1;" PREMI UN TASTO
      PER CONTINUARE "
7500 GO TO 9999
9005 RESTORE 9020
9010 FOR n=1 TO 10: READ a$:
      FOR m=0 TO 7: READ a:
      POKE USR a$+m,a: NEXT m:
      NEXT n
9020 DATA "a",24,60,189,189,255,
      189,153,153
9030 DATA "b",195,135,31,63,127,
      127,255,255
9040 DATA "c",0,7,24,32,32,64,12
      8,128
9050 DATA "d",0,224,24,4,4,2,1,1
9060 DATA "e",1,1,6,4,8,16,224,0

```

```

9070 DATA "f",128,128,96,32,16,8
      ,7,0
9080 DATA "g",128,192,224,240,24
      8,252,254,255
9090 DATA "h",1,3,7,15,31,63,127
      ,255
9100 DATA "i",24,52,52,24,126,24
      ,36,54
9110 DATA "j",24,24,60,60,126,12
      6,255,255
9150 RETURN
9500 REM istruzioni
9510 BORDER 1: INK 0: PAPER 7:
      BRIGHT 1: CLS
9530 PRINT AT 1,0; INK 1;z$;
      INK 0;'"La Terra ha quasi
      esaurito le scorte di mi
      nerali e carburanti.Il tuo
      compito e' quindi quello di
      atterrare sulla Luna e cre
      are una base per esplorar
      e la superficie."
9535 PRINT '"La missione non e'
      semplice : infatti il pi
      aneta e' circondato da una f
      ascia di pericolosissimi as
      teroidi, mentre l'allunaggio
      deve essere effettuato pu
      ntando una piattaforma gall
      eggiante nel Mare della
      Tranquillita'."
9540 PRINT #1; PAPER 1; INK 7;
      BRIGHT 1; FLASH 1;" PREMI
      UN TASTO PER CONTINUARE "

```

```

9550 PAUSE 0: CLS
9560 PRINT AT 1,0; INK 1;z$
9570 PRINT '"Gli ostacoli sono:"
      '"1/LA FASCIA DI ASTEROIDI
      '"Evita gli asteroidi alla
      deriva'"2/IL MARE'"At
      terra sulla piattaforma'"
      3/LA SUPERFICIE LUNARE'"Ev
      ita ad ogni costo i crateri
      , mentre cammini sulla sup
      erficie."
9575 PRINT AT 15,10;"MOVIMENTO"
      "Nelle prime due fasi:""5.
      ....sinistra      8.....des
      tra'"Nell'ultima fase:""O
      gni tasto ti fa saltare"
9580 PRINT #1; PAPER 1; INK 7;
      BRIGHT 1; FLASH 1;" PREMI
      UN TASTO PER CONTINUARE "
9590 PAUSE 0: CLS : RETURN
9998 FOR n=0 TO 21: PRINT AT n,0
      ; PAPER 7; INK 2; OVER 1;"
      ": NEXT n: PRINT AT 0,0
      ; PAPER 6; INK 0;"Non hai c
      ompletato la missione !":
      BEEP 1,-20
9999 PAUSE 0: PAPER 6: CLS :
      PRINT AT 1,0; INK 1;z$;
      AT 15,3; INK 0; FLASH 1;"UN
      ALTRO TENTATIVO ? (S/N)":
      PAUSE 0: IF INKEY$="s"
      THEN GO TO 9

```

Squalo!



16'48K

Ecco finalmente un programma ispirato al mare. Siete un ottimo nuotatore e vi aggirate per gli oceani alla ricerca dei favolosi tesori sommersi; tuttavia il pericolo è sempre in agguato: si tratta dei voracissimi squali che infestano i mari di tutto il mondo. Inutile dire che dovrete dunque cercare di raccogliere il maggior numero possibile di tesori, prima di essere azzeccati per tre volte da queste poderose macchine di morte. Evitate comunque anche i più subdoli granchi, innocui in apparenza ma in

realtà velenosissimi; e tenete presente che il fondo marino tenderà per tutta la durata del gioco a salire verso l'alto riducendo così le vostre possibilità di manovra. Per affrontare tutti questi ostacoli potete solo contare sulla vostra abilità di nuotatori; usate dunque i tasti '6' & '7' per nuotare verso l'alto o verso il basso; e il tasto '5' per nuotare invece in avanti a velocità doppia. Sfruttate inoltre il fatto che il programma vi consente di uscire dallo schermo in alto e in basso: altrimenti difficilmente

potrete sfuggire alle affilate dentature degli | squali!

```
1 REM *** SQUALI ***
30 GO SUB 6000
40 LET hs=0: LET n$=""
50 LET s=1: LET z=22: LET u=0:
  LET tes=0: LET vite=3:
  LET char=164: LET yd=0:
  LET xd=0
60 LET x=INT (RND*13): LET y=
  INT (RND*(z-3))+1
70 LET x1=INT (RND*12)+18:
  LET y1=INT (RND*(z-3))+1
80 GO SUB 3000: GO SUB 1000
100 LET u=u+1: IF u=35 THEN
  GO SUB 3005: RESTORE :
  GO SUB 1000: LET u=0
110 IF RND<.22 THEN GO SUB 200
  0
120 IF RND<.07 THEN GO SUB 210
  0
130 FOR t=1 TO 2
140 LET xd=-1
150 LET a$=INKEY$
160 IF a$="5" THEN LET xd=xd-1

170 IF a$="7" THEN LET yd=-1
180 IF a$="6" THEN LET yd=1
190 LET y1=y1+yd: LET x1=x1+xd
200 IF ATTR (y1,x1)=171 OR
  ATTR (y1,x1+1)=171 OR
  ATTR (y1+1,x1)=171 OR
  ATTR (y1+1,x1+1)=171 THEN
  GO SUB 4000
210 IF ATTR (y1,x1)=41 OR
  ATTR (y1,x1+1)=41 OR ATTR (
  y1+1,x1)=41 OR ATTR (y1+1,x
  1+1)=41 THEN GO SUB 5000
220 GO SUB 800
230 IF x1<0 THEN LET x1=30
240 IF y1=z-1 THEN LET y1=1
250 IF y1=0 THEN LET y1=z-2
260 LET yd=0: LET xd=0
270 IF s=1 THEN GO SUB 700:
  LET s=2: GO TO 290
280 IF s=2 THEN PRINT AT y1,x1
  ; INK 2;"RS";AT y1+1,x1;"TU
  ": LET s=1
290 NEXT t
300 PRINT AT y,x;" ";AT y+1
  ,x;" ";AT y+2,x;" "
310 LET x=x+1
320 IF y1<y AND y>1 THEN LET y
  =y-1
330 IF y1>y AND y<z-3 THEN
  LET y=y+1
```

```
340 IF x=27 THEN LET x=0
350 GO SUB 600
360 GO TO 100
500 REM subroutines
599 REM squali
600 PRINT AT y,x; INK 1;"A BCD"
  ;AT y+1,x;"EFGHIJ";AT y+2,x
  ;"'LM": RETURN
699 REM uomo
700 PRINT AT y1,x1; INK 2;"NO";
  AT y1+1,x1;"PQ": RETURN
799 REM cancella uomo
800 PRINT AT y1-yd,x1-xd;" ";
  AT y1-yd+1,x1-xd;" ":
  RETURN
999 REM UDG
1000 FOR a=144 TO char: FOR b=0
  TO 7: READ c: POKE USR
  CHR$ a+b,c: NEXT b: NEXT a
1010 LET char=145: RETURN
1020 DATA 0,0,64,32,32,16,24,12
1030 DATA 0,0,0,0,0,0,3,127
1040 DATA 0,8,12,14,15,15,255,25
  5
1050 DATA 0,0,0,0,0,128,224,255
1060 DATA 12,15,7,7,7,7,6,4
1070 DATA 3,71,255,255,255,255,3
  ,2
1080 DATA 255,255,255,255,255,25
  5,255,192
1090 DATA 255,255,255,255,255,25
  5,255,127
1100 DATA 255,255,255,255,255,25
  5,255,240
1110 DATA 224,252,223,254,248,24
  0,128,0
1120 DATA 4,8,16,0,0,0,0,0
1130 DATA 7,7,7,6,6,4,0,0
1140 DATA 240,240,224,224,224,19
  2,128,0
1150 DATA 0,7,7,6,6,118,118,127
1160 DATA 0,240,240,0,15,15,12,2
  52
1170 DATA 127,6,6,254,254,0,0,0
1180 DATA 252,12,15,15,0,0,0,0
1190 DATA 0,0,254,254,6,118,118,
  127
1200 DATA 0,0,0,0,15,15,12,252
1210 DATA 127,6,6,7,7,0,0,0
1220 DATA 252,12,15,255,252,0,0,
  0
1999 REM tesoro
2000 PRINT AT INT (RND*(z-3))+2,
  INT (RND*29)+1; INK 3;
  FLASH 1;"*"
```

```

2010 BEEP .15,30: BEEP .03,35
2020 RETURN
2099 REM granchi
2100 RESTORE 2110
2102 FOR f=USR "a" TO USR "b"+7
2104 READ c: POKE f,c
2106 NEXT f
2110 DATA 51,15,111,159,47,69,32
      ,24
2115 DATA 204,240,246,249,244,16
      2,4,24
2120 PRINT AT INT (RND*(z-4))+2,
      INT (RND*28); INK 1;"AB"
2130 BEEP .15,-20: BEEP .05,-40
2140 RESTORE 1020: GO SUB 1000
2150 RETURN

2999 REM allestimento
3000 BORDER 1: PAPER 5: CLS
3010 PRINT AT 0,2; INK 0;"Tesoro
      :"; INK 3; BRIGHT 1;tes;
      AT 0,23; INK 0; BRIGHT 0;"v
      ite:"; INK 3; BRIGHT 1;vite

3020 IF z=8 THEN LET z=21:
      GO TO 3000
3030 LET z=z-1: FOR b=0 TO 7:
      POKE USR "a"+b,INT (RND*25
      5)+1: NEXT b
3040 FOR a=0 TO 31
3050 PRINT AT z,a; INK 6;"A"
3060 BEEP .01,INT (RND*24)
3070 NEXT a
3080 IF z=y+2 THEN LET y=y-1
3090 IF z=y1+1 THEN LET y1=y1-1

3100 RETURN
3999 REM tesoro preso
4000 GO SUB 800: GO SUB 700:
      FOR a=1 TO 6: BEEP .03,a+1
      0+(21-z): NEXT a
4010 LET tes=tes+1*(22-z)
4020 PRINT AT 0,9; BRIGHT 1;
      INK 3;tes
4030 RETURN
4999 REM divorato dagli squali
5000 GO SUB 800: FLASH 1:
      BRIGHT 1: GO SUB 700:
      FLASH 0: BRIGHT 0: FOR a=0
      TO -60 STEP -1: BEEP .03,a
      : BORDER INT (RND*8):
      NEXT a
5010 LET vite=vite-1: PRINT AT 0
      ,28; INK 3; BRIGHT 1;vite
5020 IF vite=0 THEN GO TO 5060
5030 RESTORE : LET z=22: LET u=0

5040 GO TO 60
5050 GO SUB 1000

```

```

5060 BRIGHT 1: FLASH 1: FOR y=1
      TO 21 STEP 3: FOR x=1 TO 3
      0 STEP 6: GO SUB 600:
      NEXT x: NEXT y
5070 PAUSE 150
5080 FLASH 0: BRIGHT 0: CLS
5090 PRINT AT 9,0;"Hai totalizza
      to ";tes;" punti"
5100 IF tes>hs THEN LET hs=tes:
      PRINT AT 11,0;"E' il migli
      or punteggio di oggi!Prego,
      inserisci il tuo nome":
      INPUT n$: GO TO 5130
5110 PRINT AT 11,0;"Il miglior p
      untteggio di oggi e':"hs;"
      , realizzato da ";n$
5130 PRINT AT 19,3;"PREMI UN TAS
      TO PER GIOCARE"
5140 RESTORE
5150 PAUSE 0: GO TO 50
5999 REM istruzioni
6000 BORDER 4: PAPER 4: INK 0:
      CLS
6010 PRINT AT 0,9;"{SG3}{G3}
      {SG3}SQUALI{SG3}{G3}{SG3}"
6020 PRINT AT 2,0;"Devi evitare
      le voraci mascelle dei terr
      ibili squali mangiatori di
      uomini. Tu sei un simpatico
      omino, immerso nell' oce
      ano in cerca di tesori (";
      FLASH 1;"*"; FLASH 0;"") c
      he""appaiono casualmente q
      ui e la'. A parte gli squal
      i, ti segnalo inoltre i gr
      anchi (da evitare!) e il fa
      tto che il fondo marino te
      nde a salire, lasciandoti p
      ocospazio per la manovra !"

6030 PRINT AT 15,0;"I Tasti sono
      :";AT 16,0;"AVANTI VELOCE..
      ..5";AT 17,0;"GIU'.....
      .....6";AT 18,0;"SU.....
      .....7"
6040 PRINT AT 20,2;"PREMI UN TAS
      TO PER INIZIARE": PAUSE 0:
      RETURN

```

**Aspettiamo
i tuoi lavori
migliori!**

Taxi

In questo gioco sei un tassista che deve guidare per tutta la città, raccogliendo i passeggeri e portandoli alla destinazione desiderata. La vittoria si ottiene guadagnando almeno 200 dollari, in modo da poter comprare il taxi per mettersi in proprio.

Ci sono due modi per procurarsi i clienti: attendere una chiamata dalla radio (che compare come messaggio nella parte bassa dello schermo) oppure raccogliere uno degli omini che compaiono qua e là nelle strade (puoi prenderli solo se il taxi è libero e non hai chiamate). Le destinazioni appaiono sulla mappa mediante la loro lettera iniziale (S per le scuole, A per l'aeroporto, ecc.). Per ogni corsa guadagni una somma proporzionale alla distanza percorsa ed una mancia; ricordati però che i soldi che hai con te vengono conteggiati solo quando li porti al deposito (in alto a sinistra nella mappa). Il movimento è comandato tramite il Joystick (Porta n. 2), ed è solo orizzontale e verticale; l'arrivo a destinazione e la rac-

colta dell'omino possono però avvenire anche in diagonale.

Il mestiere non è privo di insidie: guardati dai tram, che si muovono lungo le strade principali e che, appena avvistano un taxi, cercano di travolgerlo. Inoltre i semafori ti lasciano passare solo quando sono verdi, mentre ti bloccano se sono gialli o rossi. Infine controlla ogni tanto, premendo il pulsante del Joystick, il livello della benzina: puoi fare il pieno quando vuoi portandoti presso uno dei distributori. Se non lo fai e resti a secco, rimani bloccato per qualche secondo (il tempo di fare un rifornimento volante di mezzo serbatoio), e in questo tempo non puoi sfuggire all'eventuale "carica" di un tram.

Si gioca a due livelli di difficoltà: Col più semplice, se la destinazione è, ad esempio, un teatro, puoi andare ad un teatro qualsiasi; col più complesso invece il computer seleziona ogni volta una meta particolare.

```

100 POKE53281,14:POKE53280,14:PRINTCHR$(142):POKE52,56:POKE56,56:CLR
110 PRINT"{CLR}":DIMA%(26),F$(26),TL%(12),T1%(12),LC%(70),CM%(26)
120 POKE214,12:PRINT:POKE211,12:PRINT"{WHT}ATTENDERE PREGO"
130 FORI=1TO26:READF$(I):NEXT
140 POKE56334,PEEK(56334)AND254:POKE1,PEEK(1)AND251
150 FORI=0TO511:POKEI+14336,PEEK(I+53248):POKEI+15360,PEEK(I+54272):NEXT
160 POKE1,PEEK(1)OR4:POKE56334,PEEK(56334)OR1
170 FORI=1TO13:READZ:FORJ=ZTOZ+7:READK:POKEJ,K:NEXTJ,I
180 FORI=1TO26:READA%(I):NEXT:FORI=1TO8:READV(I):NEXT
190 FORI=1TO26:IFI=1THENCM%(I)=A%(I):GOTO210

```

```

200 CM%(I)=CM%(I-1)+A%(I)
210 NEXT:H=1024:C=54272:S=C:CH=28:FORI=1TO5:READB%(I),E%(I):NEXT
220 FORI=1TO12:READT1%(I):NEXT
230 DEFFNA(X)=INT(RND(1)*X)+1:DEFFNL(Q)=H+40*Y+X
240 DEFFND(ZZ)=ABS((ZZ=39)+(ZZ=-39)+(ZZ=41)+(ZZ=-41))
250 FORI=1TO5:A=FNA(2):D%(I)=(A=1)-(A=2):NEXT
260 FORI=1TO5:TR(I)=(D%(I)=-1)*(-E%(I))-(D%(I)=1)*(B%(I))
270 VR(I)=(TR(I)=E%(I))-(TR(I)=B%(I)):IFI>2THENVR(I)=VR(I)*40
280 TY=(VR(I)=1)*36+(VR(I)=-1)*37+(VR(I)=40)*39+(VR(I)=-40)*38:TY(I)=(-1)*TY
290 ED(I)=(TR(I)=B%(I))*(-E%(I))+(TR(I)=E%(I))*(-B%(I)):NEXT

```



```

300 PRINT "{CLR}":POKE214,8:PRINT
:POKE211,9:PRINT "{WHT}SELEZI
ONARE IL LIVELLO"
310 POKE214,10:PRINT:POKE211,14:
PRINT"1) FACILE"
320 POKE214,12:PRINT:POKE211,14:
PRINT"2) DIFFICILE"
330 GETZ$:LV=VAL(Z$):RD=RND(1):I
FZ$=""THEN330
340 IFLV<1ORLV>2THEN330
350 PRINT "{CLR}":POKE53272,(PEEK
(53272)AND240)OR14
360 FORMC=STOS+24:POKEMC,0:NEXT:
POKES+24,15:POKES+5,120:POKE
S+6,240:HF=S+1:LF=S
370 PRINT "{CLR}":GOSUB1000:FORI=
1TO12:A=TL%(I):GOSUB1170:NEX
T
380 P%=H+81:POKEP%,CH:POKEP%+C,7
390 M1=0:M2=0:M3=0:M4=0:XF%=0:CF
%=0:GS=1400
400 GOSUB800:IFEN=1THEN1810
410 IFXF%=0THENGOSUB960
420 GOSUB480:IFEN=1THEN1810
430 GOSUB1890:GOSUB550:IFEN=1THE
N1810
440 GOSUB800:IFEN=1THEN1810
450 GOSUB1570:IFEN=1THEN1810
460 GOSUB550:IFEN=1THEN1810
470 GOTO400
480 FORT=1TO5:A=TR(T):TR(T)=TR(T)
+VR(T)
490 IFTR(T)=ED(T)THENVR(T)=(-1)*
VR(T):GOSUB740:GOTO520
500 IFPEEK(TR(T))>=36ANDPEEK(TR(
T))<=39THENTR(T)=TR(T)-VR(T)
:GOTO540
510 IFPEEK(TR(T))=27ORPEEK(TR(T)
)=28THENEN=1:T=5:GOTO540
520 POKEA,32:POKETR(T),TY(T):POK
ETR(T)+C,15:GOSUB2300:IFN4=1
THEN540
530 GOSUB550:IFEN=1THENT=5
540 NEXT:RETURN
550 JV=PEEK(56320):JV=15-(JVAND1
5):FR=PEEK(56320)AND16
560 IFFR=0ANDSH=0THENGOSUB2120
570 DY=(JV=1)+(JV=5)+(JV=9)-(JV=
6)-(JV=10)-(JV=2)
580 DX=(JV=4)+(JV=5)+(JV=6)-(JV=
9)-(JV=10)-(JV=8)
590 IF(DX=0ANDDY=0)ORGS=0THENRET
URN
600 CH=28:IFDX<>0THENCH=28
610 IFDY<>0THENCH=27
620 ZZ=40*DY+DX:Z=P%+ZZ:PK=PEEK(
Z)
630 IFFND(ZZ)=1AND(PK=32ORPK=160
)THEN720
640 IFSH=1THEN770
650 CP=(PK=30)+(PK=32)*2+(PK=40)
*3+(PK=41)*4+(PK=42)*5+(PK=1
60)*6+(PK=0)*7
660 ONABS(CP)GOTO690,710,1930,19
30,2090,720,1220
670 IFPEEK(Z)>=129ANDPEEK(Z)<=15
4THEN1320
680 IFPEEK(Z)>=36ANDPEEK(Z)<=39T
HENEN=1:RETURN
690 IF(PEEK(Z+C)AND15)=5THEN710
700 GOTO720
710 GOSUB2310:POKEP%,32:POKEZ,CH
:POKEZ+C,7:P%=Z
720 GS=GS-2:IFGS<=0THENGGS=0:GOSU
B2210
730 RETURN
740 TY=(VR(T)=1)*36+(VR(T)=-1)*3
7+(VR(T)=40)*39+(VR(T)=-40)*
38:TY(T)=(-1)*TY
750 ED=(ED(T)=B%(T))*(-E%(T))+(E
D(T)=E%(T))*(-B%(T)):ED(T)=E
D
760 RETURN
770 IFPEEK(Z)=30AND(PEEK(Z+C)AND
15)=5THEN710
780 IFPEEK(Z)<>32THEN720
790 GOTO710
800 FORLL=1TO12:PK=PEEK(TL%(LL)+
C)AND15
810 IFPEEK(TL%(LL))=32THENPOKETL
%(LL),30:POKETL%(LL)+C,PK
820 POKETL%(LL)+C,PK:TL=(PK=2)*1
+(PK=5)*2+(PK=7)*3
830 GOSUB550:IFEN=1THENLL=12:GOT
O870
840 GOSUB480:IFEN=1THENLL=12:GOT
O870
850 GOSUB1570:IFEN=1THENLL=12:GO
TO870
860 ONABS(TL)GOSUB880,900,920
870 NEXT:RETURN
880 A=FNA(3):IFA=1THENPOKETL%(LL)
+C,5:POKETL%(LL),30
890 RETURN
900 A=FNA(3):IFA=2THENPOKETL%(LL)
+C,7:POKETL%(LL),30
910 RETURN
920 A=FNA(3):IFA=3THENPOKETL%(LL)
+C,2:POKETL%(LL),30
930 RETURN
940 FORI=22TO23:FORJ=1TO37
950 POKE214,I:PRINT:POKE211,J:PR
INTCHR$(32);:NEXTJ,I:RETURN
960 PX=FNA(26):XF%=PX+128

```

```

970 GOSUB940:POKE214,22:PRINT:PO
KE211,1
980 PRINT"{WHT}TAXI, VAI AL";F$(
XF%-128);:PT=PX:CL=0:X=1:GOS
UB1760
990 GOSUB550:RETURN
1000 I=49152:IFPEEK(49154)=216TH
ENSYS49160:GOTO1030
1010 READA:IFA=256THENSYS49160:G
OTO1030
1020 POKEI,A:I=I+1:GOTO1010
1030 POKE1064,40:POKE1065,41:POK
E1066,40:POKE1067,41:FORI=1
064TO1067:POKEI+C,1
1040 NEXT:POKE1104+C,1:POKE1104,
40:FORI=1105TO1107:POKEI,32
:POKEI+40,32:NEXT
1050 POKE1144,40:POKE1144+C,1:FO
RI=1265TO1301:POKEI,32:NEXT
1060 FORI=1545TO1581:POKEI,32:NE
XT
1070 FORI=1114TO1754STEP40:POKEI
,32:POKEI+10,32:POKEI+20,32
:NEXT
1080 FORI=1TO5
1090 X=FNA(38):Y=FNA(19):L=FNL(Q
)
1100 IFPEEK(L)=32ORPEEK(L)<>160T
HEN1090
1110 POKEL,42:POKEL+C,7:NEXT
1120 K=0:FORI=1TO26:FORJ=1TOA%(I
)
1130 X=FNA(40):Y=FNA(19):L=FNL(Q
)
1140 IFPEEK(L)=32ORPEEK(L)<>160T
HEN1130
1150 IFPEEK(L+1)=160ANDPEEK(L-1)
=160ANDPEEK(L+40)=160ANDPEE
K(L-40)=160THEN1130
1160 K=K+1:LC%(K)=L:POKEL,I+128:
NEXT:NEXT:RETURN
1170 IFPEEK(A)<>32THENGOSUB1190
1180 TL%(I)=A:POKEA,30:POKEA+C,5
:RETURN
1190 FORJ=1TO8
1200 IFPEEK(A+V(J))=32THENA=A+V(
J):K=8
1210 NEXT:RETURN
1220 IFXF%<>0THENRETURN
1230 POKEP%,32:POKEZ,CH:POKEZ+C,
7:GOSUB960
1240 POKEP%,CH:POKEP%+C,7:POKEZ,
160:POKEZ+C,1:MN=1:RETURN
1250 M1=ABS(SP-Z)/10+1:POKEP%,32
:POKEZ,CH:POKEZ+C,7:GOSUB23
10
1260 M2=INT(FNA(100*M1)/10)/100

```

```

1270 GOSUB940:POKE214,22:PRINT:P
OKE211,1
1280 PRINT"HAI GUADAGNATO";M1;"E
";M2;"DI MANCIA";
1290 M3=M1+M2+M3:M1=0:M2=0:POKEP
%,CH:POKEP%+C,7:POKEZ,XF%:P
OKEZ+C,1
1300 PT=PX:CL=1:X=1:GOSUB1760
1310 POKE214,23:PRINT:POKE211,1:
PRINT"HAI CON TE";M3;:MN=0:
XF%=0:CF%=0:RETURN
1320 IFMN=1THEN1250
1330 IFCF%<>0THEN1450
1340 IFPEEK(Z)<>XF%THEN730
1350 IFLV=2ANDZ<>DZTHEN730
1360 PC=FNA(26):CF%=PC+128
1370 SP=Z:POKEZ,CH:POKEZ+C,7:POK
EP%,32:GOSUB2310
1380 IFLV=1THENPT=PC:CL=1:GOSUB1
760
1390 GOSUB940:POKE214,22:PRINT:P
OKE211,1
1400 PRINT"{WHT}PORTAMI AL";F$(C
F%-128);:PT=PC:CL=0:X=2:GOS
UB1760
1410 POKEP%,CH:POKEP%+C,7:GOSUB2
310:POKEZ,XF%:POKEZ+C,1
1420 IFCF%=XF%THENPOKEZ+C,0
1430 IFLV=1ANDXF%<>CF%THENPT=PX:
CL=1:GOSUB1760
1440 GOTO730
1450 IFPEEK(Z)<>CF%THEN730
1460 IFLV=2ANDZ<>DZTHEN730
1470 POKEZ,CH:POKEZ+C,7:POKEP%,3
2:GOSUB2310
1480 M1=ABS(SP-Z)/10+1

```



```

1490 M2=INT(FNA(100*M1)/10)/100
1500 GOSUB940:POKE214,22:PRINT:P
OKE211,1
1510 PRINT"HAI GUADAGNATO";M1;"E
";M2;"DI MANCIA";
1520 M3=M1+M2+M3:M1=0:M2=0
1530 POKE214,23:PRINT:POKE211,1:
PRINT"HAI CON TE";M3;
1540 POKEP%,CH:POKEP%+C,7:GOSUB2
310:POKEZ,CF%:POKEZ+C,1
1550 IFLV=1THENPT=PC:CL=1:GOSUB1
760
1560 CF%=0:XF%=0:GOTO730
1570 MX=0:MY=0:GOTO1590
1580 MY=INT((P%-H)/40):MX=(P%-H)
-40*MY:RETURN
1590 GOSUB1580:CT=ABS((MY=6)*1+(
MY=13)*2):T=CT:IFCT>0THENGOS
SUB1620
1600 GOSUB1580:CT=ABS((MX=10)*3+
(MX=20)*4+(MX=30)*5):T=CT:IF
CT>0THENGOSUB1620
1610 RETURN
1620 IFP%>TR(CT)ANDVR(CT)>0THEN1
650
1630 IFP%<TR(CT)ANDVR(CT)<0THEN1
650
1640 VR(CT)=(-1)*VR(CT):GOSUB740
1650 B=FNA(2):DN=(B=1)*(-P%)-(B=
2)*(ED(CT))
1660 POKETR(CT),32:FORMV=TR(CT)T
ODNSTEPVR(CT):PK=PEEK(MV)
1670 IFPK>=36ANDPK<=39THENTR(CT)
=MV-VR(CT):MV=DN:NEXT:GOSUB
1750:RETURN
1680 IFPK=27ORPK=28THENMV=DN:NEX
T:EN=1:RETURN
1690 POKEMV,TY(CT):POKEMV+C,15:S
H=1
1700 GOSUB2300:GOSUB550:SH=0:POK
EMV,32
1710 IFEN=1THENMV=DN
1720 NEXT:IFEN=1THENRETURN
1730 IFDN=ED(CT)THENTR(CT)=ED(CT)
-VR(CT):POKEED(CT),TY(CT):
RETURN
1740 TR(CT)=DN:POKETR(CT),TY(CT)
:RETURN
1750 POKETR(CT),TY(CT):GOSUB550:
RETURN
1760 IFLV=2THENONXGOTO1790,1800
1770 FORLC=CM%(PT-1)+1TOCM%(PT)
1780 POKELC%(LC)+C,CL:NEXT:RETUR
N
1790 R=FNA(A%(PT)):DZ=LC%(CM%(PT
-1)+R):POKEDZ+C,0:RETURN
1800 R=FNA(A%(PT)):DC=LC%(CM%(PT
-1)+R):POKEDC+C,0:RETURN
1810 PRINT"{CLR}":EN=0:GOSUB2320
:POKE53272,21
1820 POKE214,8:PRINT:POKE211,6:P
RINT"{WHT}SEI STATO INVESTI
TO DA UN TRAM"
1830 POKE214,10:PRINT:POKE211,14
:PRINT"{RVS} F I N E !
{OFF}"
1840 POKE214,12:PRINT:PRINT" HAI
GUADAGNATO IN TUTTO {RVS}"
;M4;"{OFF}DOLLARI"
1850 PRINT"{ 2 GIU' }{ 10 DES}GIO
CHI ANCORA (S/N)"
1860 GETZ$:RD=RND(1):IFZ$=""OR(Z
$<>"S"ANDZ$<>"N")THEN1860
1870 IFZ$="N"THENSYS2048
1880 PRINT"{CLR}":RESTORE:GOTO23
0
1890 X=FNA(40):Y=FNA(19)
1900 IFPEEK(FNL(Q))<>160THEN1920
1910 POKEFNL(Q),0:POKEFNL(Q)+C,3
1920 RETURN
1930 M4=M4+M3:GOSUB2310
1940 GOSUB940:POKE214,22:PRINT:P
OKE211,1:PRINT"{WHT}AL SICU
RO NEL DEPOSITO ";M4
1950 M3=0:POKEP%,CH:POKEP%+C,7:P
OKEZ,40:POKEZ+C,1:FORDL=1TO
500:NEXT
1960 IFM4>200THEN2050
1970 IFCF%<>0THEN2020
1980 IFFX%=0THEN730
1990 GOSUB940:POKE214,22:PRINT:P
OKE211,1
2000 PRINT"{WHT}TAXI, VAI AL";F$
(XF%-128);:N4=1:GOSUB480:N4
=0
2010 RETURN
2020 GOSUB940:POKE214,22:PRINT:P
OKE211,1

```

PAPER
24 soft

ritaglia
e
conserva
ti
attendono
grosse
sorprese!


```

2030 PRINT"{WHT}MI PORTI AL";F$(
CF%-128);:N4=1:GOSUB480:N4=
0
2040 RETURN
2050 PRINT"{CLR}":POKE214,8:PRIN
T:POKE211,7:PRINT"{WHT}HAI
GUADAGNATO ABBASTANZA,"
2060 POKE214,10:PRINT:POKE211,13
:PRINT"IL TAXI EK TUO"
2070 PRINT"{ 2 GIU' }{ 9 DES}
{RVS} H A I{ 3 SPAZI}V I N
T O ! {OFF}":GOSUB2340
2080 POKE53272,21:GOTO1850
2090 POKEZ,CH:POKEZ+C,15:POKEP%,
32:GS=1400
2100 POKEP%,CH:POKEP%+C,1:POKEZ,
42:POKEZ+C,7
2110 RETURN
2120 GOSUB940:POKE214,22:PRINT:P
OKE211,1
2130 PRINT"{WHT}{RVS} {OFF}V
{ 5 SPAZI}P{RVS} {OFF}"
2140 A1=INT(GS/200):POKE214,23:P
RINT:POKE211,A1+1:PRINT"
{CYN}"CHR$(95);
2150 FORDL=1TO500:NEXT:IFCF%=0TH
EN2180
2160 GOSUB940:POKE214,22:PRINT:P
OKE211,1
2170 PRINT"{WHT}PORTAMI AL";F$(C
F%-128);:N4=1:GOSUB480:N4=0
:RETURN
2180 IFXF%=0THENRETURN
2190 GOSUB940:POKE214,22:PRINT:P
OKE211,1
2200 PRINT"{WHT}TAXI,VAI AL";F$(
XF%-128);:N4=1:GOSUB480:N4=
0:RETURN
2210 GOSUB940:POKE214,22:PRINT:P
OKE211,1
2220 PRINT"{WHT}E' FINITA LA BEN
ZINA":N4=1:GOSUB480:N4=0:IF
EN=1THENRETURN
2230 GOSUB940:POKE214,22:PRINT:P
OKE211,1
2240 PRINT"ORA HAI MEZZO SERBATO
IO PIENO{WHT}":N4=1:GOSUB4
80:N4=0:IFEN=1THENRETURN
2250 GOSUB1570:IFEN=1THENRETURN
2260 GS=700:GOSUB940:POKE214,22:
PRINT:POKE211,1
2270 IFXF%<>0THENPRINT"{WHT}TAXI
, VAI AL";F$(XF%-128);:RETU
RN
2280 IFCF%<>0THENPRINT"{WHT}PORT
AMI AL";F$(CF%-128);:RETURN
2290 RETURN
2300 POKES+4,17:POKEHF,40:POKELF
,250:POKES+4,16:RETURN
2310 POKES+4,33:POKEHF,50:POKELF
,100:POKES+4,32:RETURN
2320 POKES+4,33:FORI=200TO70STEP
-5:POKEHF,I:POKELF,INT(I/2)
:NEXT:POKES+4,32
2330 POKES+24,0:RETURN
2340 POKES+4,17:POKEHF,40:POKELF
,200:FORI=1TO10:FORJ=70TO20
0STEP5:POKEHF,J
2350 POKELF,90:NEXT:G=G+1:IFG<5T
HEN2340
2360 POKES+4,16:POKES+24,0:RETUR
N
2370 DATA"LKAEROPORTO","LA BANCA
"," CAMPO SPORTIVO","LA DRO
GHERIA"
2380 DATA"LKESATTORIA","LA FABBR
ICA","LA GIOIELLERIA","LKHO
TEL","LKIPPODROMO"
2390 DATA"JÄZZ CLUB","KING BAR
","LA LIBRERIA","MERCATO",
"NIGHT CLUB"
2400 DATA"LKOSPEDALE","LA POSTA"
,"LA QUÈSTURA","RISTORANTE
","LA SCUOLA"
2410 DATA"TEATRO","LKUNIVERSITA
K","LA VILLA","WHISKY BAR"
2420 DATA"LO XENON","LO YACHT CL
UB","LO ZOO"
2430 DATA14552,24,255,255,24,24,
255,255,24
2440 DATA14560,0,102,126,255,255
,126,102,0
2450 DATA14936,2,4,8,0,0,0,0,0
2460 DATA14576,0,8,42,42,42,8,4,
4
2470 DATA14584,8,28,62,28,28,28,
28,28
2480 DATA14336,60,60,25,127,124,
124,24,60
2490 DATA14624,64,32,15,255,131,
253,253,108
2500 DATA14632,2,4,8,255,193,191
,191,54
2510 DATA14640,30,24,23,23,64,87
,151,30
2520 DATA14648,30,151,87,64,23,2
3,24,30
2530 DATA14656,0,255,170,255,136
,136,136,136
2540 DATA14664,0,255,171,255,143
,143,141,141
2550 DATA14672,15,25,41,47,47,47
,31,15

```

```

2560 DATA1,5,2,3,1,2,5,1,2,4,1,1
      ,6,7,5,2,1,6,3,5,1,2,1,1,1,
      1
2570 DATA1,-1,40,-40,-39,39,-41,
      41
2580 DATA1265,1301,1545,1581,111
      4,1754,1124,1764,1134,1774
2590 DATA1151,1159,1169,1179,142
      9,1439,1449,1459,1669,1679,
      1689,1699
2600 DATA1,0,216,255,255,255,40
2610 DATA0,169,81,133,251,169,40

2620 DATA133,253,169,4,133,252,1
      33
2630 DATA254,169,147,32,210,255,
      162
2640 DATA0,160,0,169,160,145,253
2650 DATA200,192,39,208,249,24,1
      65
2660 DATA253,105,40,133,253,144,
      2
2670 DATA230,254,232,224,19,208,
      229
2680 DATA160,0,169,4,145,251,169
2690 DATA255,141,15,212,169,128,
      141
2700 DATA18,212,173,27,212,41,3
2710 DATA133,173,170,10,168,24,1
      85

```

```

2720 DATA0,192,101,251,133,170,1
      85
2730 DATA1,192,101,252,133,171,2
      4
2740 DATA185,0,192,101,170,133,2
      53
2750 DATA185,1,192,101,171,133,2
      54
2760 DATA160,0,177,253,201,160,2
      08
2770 DATA18,138,145,253,169,32,1
      45
2780 DATA170,165,253,133,251,165
      ,254
2790 DATA133,252,76,62,192,232,1
      38
2800 DATA41,3,197,173,208,189,17
      7
2810 DATA251,170,169,32,145,251,
      224
2820 DATA4,240,26,138,10,168,162

2830 DATA2,56,165,251,249,0,192
2840 DATA133,251,165,252,249,1,1
      92
2850 DATA133,252,202,208,238,76,
      62
2860 DATA192,169,1,160,0,153,0
2870 DATA216,153,0,217,153,0,218

2880 DATA153,0,219,200,208,241,9
      6,256

```

RICERCA COLLABORATORI

Il Gruppo Editoriale Jackson nel quadro di potenziamento delle attività editoriali ricerca collaboratori esperti nella programmazione di Commodore 64, VIC 20 e ZX Spectrum.

Requisiti indispensabili sono:

- disponibilità del computer da almeno 8 mesi*
- predisposizione alla grafica e suono*
- residenza in Lombardia*

*Per maggiori informazioni telefonare a Gruppo Editoriale Jackson
Tel. 02/6880951-2-3-4-5 chiedendo della Sig.na Renata Rossi*

ERRATA CORRIGE

Nel programma "Gestione Files Dati" per TI 99 pubblicato su numero 19 della rivista, la linea 330 è errata e va corretta nel modo seguente:

330 NEXT I :: GOTO 290

ERRATA CORRIGE

*Nel programma "Diagrammi a torta su printer plotter I520" pubblicato su Paper-soft N. 19, per un guaio della stampante sono "saltati" tutti i caratteri di Pi greco (π). Le linee 40,400,410,448,1010,1015,2005,2010,2410,2605,3010,3020,3120 e 3130, che contengono i caratteri */ vanno quindi modificate aggiungendo il carattere di Pi greco: * π /*

Il coniglietto

Sei alla guida del simpatico Bunny, che è molto affamato, e devi aiutarlo perciò a raggiungere e mangiare le carote che sono il suo cibo preferito e che, sfortunatamente, si trovano sospese nell'aria. In compenso il tuo amico è per sua natura un abile saltatore e, per sfruttare questa sua capacità, dovrai utilizzare il Joystick che ti permetterà, premendo il pulsante di fuoco o muovendo in avanti la leva, di fargli spiccare un balzo e, sempre muovendo la leva a destra e a sinistra, di dirigerlo durante il suo volo.

Il coniglietto può servirsi delle piattaforme fisse ai lati dello schermo o delle nuvole che si muovono nel cielo per raggiungere le sue amate carote che ti fruttano 20 punti ognuna.

Bunny però non è un coniglio infaticabile, quindi devi cercare di recuperare tutte le carote di ogni

schermata in meno di 60 secondi; se dopo 30 secondi dalla partenza non avrai ancora divorato tutte le carote presenti sullo schermo, le nuvole cominceranno a muoversi più velocemente e, se allo scadere dei 60 secondi non avrai ancora compiuto la tua missione, le nuvole inizieranno a scomparire.

Inoltre devi stare molto attento a non far cadere il tuo amico sul prato perché ogni volta che ciò accade, egli perde una delle sue tre vite. Se riuscirai a mangiare tutte le carote presenti nello schermo, potrai passare al livello successivo.

Ogni cinque livelli, ce n'è uno nel quale il coniglietto salta continuamente.

Inoltre, ogni dieci livelli superati, il coniglietto acquisterà una vita in più.

```
5 DATA0,0,0,0,0,0,0,0,8,12,6,189
  ,127,124,56,70,16,48,96,189,25
  4,62,28,34
10 DATA0,102,255,255,102,0,0,0,6
  6,40,134,51,136,102,102,102
15 DATA0,48,188,255,255,188,48,0
  ,0,102,255,255,102,0,0,0
20 FORX=7424TO7479:READY:POKEX,Y
  :NEXT:S=36876:I=100
25 POKE36869,255:POKES+3,109:K=3
  0720:L=10:CT=4:LI=3:GOSUB600:
  POKES+2,15
30 DATA160,22,177,253,200,145,25
  3,136,136,192,255,208,245,160
  ,22,177,253,160,0,145
35 DATA253,160,22,169,32,145,253
  ,96,160,23,177,253,136,145,25
  3,200,200,192,45,208
40 DATA245,160,22,177,253,160,44
  ,145,253,160,22,169,32,145,25
  3,96,169,30,133,254,169
45 DATA153,133,253,32,184,28,169
  ,220,133,253,32,156,28,169,29
  ,133,253,169,31,133
50 DATA254,32,184,28,169,96,133,
  253,32,156,28,169,161,133,253
  ,32,184,28,96
```

```
55 FORX=7324TO7423:READY:POKEX,Y
  :NEXT
100 PRINT"{HOME}{ 4 SPAZI}{YEL}
  {RVS}IL CONIGLIETTO":PRINT"
  {WHT}{RVS}TEMPO":PRINT"
  {HOME}{ 2 GIU' }{ 5 SIN }{RVS}
  PUNTI"
105 FORX=8164TO8185:POKEX,36:POK
  EX+K,5:NEXT
110 FORX=38576TO38861:POKEX,1:NE
  XT:I=I-3

115 PRINT"{HOME}{ 6 GIU' }{RVS}
  {YEL}{<Y>}{ 9 DES }{< 2 Y>}
  { 9 DES }{<Y>}{ 5 GIU' }
  {< 2 Y>}{ 4 SIN }{GIU' }
  {< 2 Y>}{ 5 GIU' }{< 2 Y>}
  { 4 SIN }{GIU' }{< 2 Y>}"
120 FORX=8TO20STEP6:FORY=1TOL:PO
  KE7680+X*22+RND(1)*22,35:NEX
  TY,X
125 FORX=10TO16STEP6:FORY=1TOL:P
  OKE7680+X*22+RND(1)*22,38:NE
  XTY,X
130 C=CT:FORX=1TOC
135 Y=INT(RND(1)*351)+7768:IFPEE
  K(Y)<>32ORY=8055THEN135
```




```

140 IF (Y<8143ANDY>8118) OR (Y<8056
ANDY>8030) OR (Y<8011ANDY>7986
) THEN 135
145 IF (Y<7879ANDY>7854) OR (Y<7924
ANDY>7898) THEN 135
150 POKEY,37:POKEY+K,7:NEXTX:P=8
055:D=33:CT=CT+1

155 TI$="000000":POKES-1,130
200 POKEP,32:POKE37154,127:Y=PEE
K(37152):IF (YAND128)=0 THEN P=
P+1:D=33
205 IF PEEK(P+22)=38 AND (NOT(B) OR Z
) THEN P=P+1
210 POKE37154,255:Y=PEEK(37137):
IF (YAND16)=0 THEN P=P-1:D=34
215 IF PEEK(P+22)=35 AND (NOT(B) OR Z
) THEN P=P-1
220 B=(B=0):IF BORZ THEN SYS7380:PO
KES-1,PEEK(S-1)+20:IF PEEK(S-
1)=170 THEN POKES-1,130
225 IFA=0 AND PEEK(P+22)<>32 THEN IF
(YAND32)=0 OR (YAND4)=0 OR V THEN
A=7-W*2:POKES,195
230 IFA<4-W AND PEEK(P+22)=32 OR PEE
K(P+22)=37 THEN P=P+22
235 IFA<>0 THEN A=A-1:IFA>3-W THEN P
=P-22:POKES,PEEK(S)+20
240 IF PEEK(P)=37 THEN 700
245 IF PEEK(P+22)=36 THEN 900
250 IF PEEK(P)<>32 THEN P=P+22:POKE
S,PEEK(S)-20:GOTO260
255 POKES,70
260 IF PEEK(P)=37 THEN 700
265 POKEP,D:POKEP+K,1:FORX=1TO1:
NEXT
270 PRINT"{HOME} { 3 GIU' }
{ 6 SIN } {RVS}";SC
275 PRINT"{HOME} { 2 GIU' } {RVS}";
:T=60-VAL(TI$):IFT<1 THEN 400
280 PRINTT;"{SIN} ":GOTO200
400 PRINT"{HOME} { 2 GIU' }
{ 2 SPAZI } {CYN} {RVS}0
{ 2 SPAZI}":Z=-1:IFTI$<"0001
30" THEN GOTO200
405 Y=7746+INT(RND(1)*417):IF PEE
K(Y)=37 OR (PEEK(Y)=32 AND RND(1
)>.5) THEN 405
410 POKEY,32:GOTO200
500 PRINT"{ 10 GIU' } { 7 DES }
{RVS} {GRN} ANCORA ?"
505 GETA$:IFA$="" THEN 505
510 IFA$="S" THEN RUN
515 IFA$="N" THEN END
520 GOTO505
600 IF LI<0 THEN 500
605 PRINT"{CLR} { 2 GIU' }
{ 9 DES } {WHT}";:IF LI>0 THEN FO
RX=1TO LI:PRINT"!";:NEXT
610 PRINT" ":LE=LE+1:Z=0:W=0:V=0
:IF LE>19 THEN W=1
615 IF INT(LE/10)=LE/10 THEN LI=LI+
1
620 IF LE/5=INT(LE/5) THEN V=-1
625 IF LE/10=INT(LE/10) THEN CT=4
630 L=L-1:IF L<3 THEN L=10
635 RETURN
650 PRINT"{CLR} { 2 GIU' }
{ 9 DES } {WHT}";:IF LI>0 THEN FO
RX=1TO LI:PRINT"!";:NEXT
700 FORX=1TO20:POKES+1,126+(20-X
)*3:SC=SC+1:PRINT"{HOME}
{ 3 GIU' } { 6 SIN } {RVS}";SC:N
EXT
705 C=C-1:ON-(C=0)+1GOTO265,800
800 POKES,0:FORC=1TOLE:Y=0:X=0:V
=0
805 Z=8142+X+Y:X=X+1:POKES-1,188
-Y:IFX=21 THEN 825
810 IFV=0 THEN Y=Y-22:IFY=-66 THEN V
=1
815 IFV=1 THEN Y=Y+22:IFY=0 THEN V=0

820 POKEZ,32:POKE8142+X+Y,33:POK
E8142+X+Y+K,1:GOTO805
825 POKEZ,32:NEXTC:POKES-1,0
830 IFT>0 THEN FORX=TTO0STEP-1:PRI
NT"{HOME} { 2 GIU' } {RVS}";X:S
C=SC+T:PRINT"{HOME}
{ 3 GIU' } { .5 SIN } {RVS}";SC
835 IFT>0 THEN POKES,0:POKES,170+T
:NEXTX
840 POKES,0:GOSUB600:GOTO100
900 FORY=1TOLE:POKES-1,0:FORX=25
5TO200STEP-1:POKES,X:NEXTX,Y
:POKES,0
905 LI=LI-1:GOSUB600:GOTO100

```



Un tocco di colore

Ecco un programma che, nonostante la sua apparente semplicità, potrà accontentare le più disparate fasce di utenza: dal lettore alle prime armi che troverà un programma dalla struttura semplice ed accessibile, di facile comprensione, verificando allo stesso tempo che poche linee di programma possono condurre a risultati interessanti; sino al programmatore già evoluto, che potrà qui trovare uno spunto per arricchire i propri programmi con colori non pre-

visti (si fa per dire!) da sir Clive.

L'idea è semplice: abbinando INK, PAPER e caratteri grafici particolari si possono ottenere tonalità di colore aggiuntive. Se lanciate il programma vedrete 6 schermi pieni di colori bellissimi: se poi identificate gli attributi di un colore a voi gradito, potrete inserirlo per colorare i vostri programmi. E pensate che, modificando il carattere grafico, potrete ottenere infinite varianti...

```
1 REM *****
2 REM * UN TOCCO DI COLORE *
3 REM *****

10 FOR n=0 TO 7
20 READ dato
30 POKE USR "a"+n,dato
40 NEXT n
```

```
50 DATA 85,170,85,170,85,170,8
5,170
60 FOR p=0 TO 7
70 FOR i=0 TO 7
80 PRINT INK i; PAPER p;"AAAA
AAAAAAAAAAAAAAAAAAAAAAAAAAAA
AAAAAAAAAAAAAAAAAAAAAAAAAAAA
AAAAAA"
90 NEXT i
100 NEXT p
110 STOP
```

Per ricevere gli arretrati di **PAPER soft** compilare il seguente tagliando:

J.soft Compilare ed inviare in busta chiusa a:
via Rosellini 12 - 20142 Milano - Tel. 02/6888228-683797-6880841-2-3

Nome _____

Cognome _____

Indirizzo _____

CAP _____ Città _____ Prov. _____


Desidero ricevere i seguenti numeri arretrati di

PAPER soft _____

a L. 2.000 cad. per un totale di L. a mezzo

☐ assegno allegato

☐ contanti allegati

**POSSIEDI
IL TI.99/4A 
E SEI ALLA
RICERCA DI
SOFTWARE?**

Approfitta dell'offerta J. soft!

In seguito alle sempre più numerose richieste di programmi da parte dei possessori dell'home computer TI-99/4A, mettiamo a disposizione degli interessati un vasto assortimento a prezzi estremamente contenuti. L'offerta è valida fino ad esaurimento delle scorte. Prima ordini, più sei sicuro di ricevere quanto desideri.



Programming Aids I - (Aiuto alla programmazione)
Cod. DTXSX04 - L. 14.000

Mette a disposizione gli ausili per potenziare il linguaggio BASIC TI. Comprende le seguenti possibilità: "CATALOGARE DISCHI", "DISPLAY AT", "ACCEPT AT", "SCREEN PRINT (per la stampa su carta dello schermo)", "LOWERCASE (per la creazione di un set di lettere minuscole)", "2nd ASCII (per la costruzione e l'utilizzo di un secondo set di caratteri)", "CHARDEF (per rea-

lizzare e facilitare la definizione dei caratteri)".

Configurazione richiesta: A
Configurazione raccomandata: D + E

Blackjack e Poker
Cod. DTXMX04 - L. 25.000

Partite di carte simulate col computer, che vi consentono di puntare le somme dei vostri desideri. Possono partecipare fino a quattro giocatori

Hunt the Wumpus (Caccia la Mostro)
Cod. DTXMA 12 - L. 25.000

Una caccia emozionante in un dedalo di caverne e gallerie. Cercate la tona del Mostro evitando i pericoli in agguato lungo il percorso. Soppesate e vogliate ottentamente gli indizi per completare questa missione pericolosa.

Configurazione raccomandata: B

Hangman Il cornefice)
Cod. DTXIX06 - L. 25.000

Il giocatore cerca di scoprire la parola segreto e ogni volta che sbaglia, si avvicina maggiormente alla forca. Potete usare 200 parole programmate in inglese oppure crearne 60 nuove.

Configurazione raccomandata: A

Othello
Cod. DTXMX20 - L. 35.000

Questo antico gioco di strategia impegna o fonda anche i giocatori più smoliziati. Un gioco da scacchiera che si impara in pochi minuti ma che richiede... una vita per dominarlo. Per strategie di tutte le età.

Tombstone City: 21° secolo
Cod. DTXMX28 - L. 25.000

Vi trovate in una città fantasma del Far West nel XXI secolo, minacciata da un'orda di invasori verdi. Avanzate con la vostra pottuglio su corri coperti tipo Far West, cercando di arrestare i verdi "morgs".
Gioco velocissimo per una persona, Tombstone City mette alla prova la Vostra abilità strategica e la prontezza di riflessi.

Configurazione raccomandata: B

Blasto
Cod. DTXMX05 - L. 25.000

Un carro armato (o due, a secondo del numero dei giocatori) entrano in lizza per far saltare un campo minato, evitando nel contempo i tiri dell'avversario. Una corsa velocissima contro il cronometro allo scopo di colpire il massimo numero di mine. Attenti a non colpire le mine se siete troppo vicini, altrimenti dovrete ricominciare da capo.

Configurazione raccomandata: B

Personal Report Generator (Generatore rapporti e archivi personali)
Cod. DTXTX01 - L. 68.000

Permette di creare, edit e stampare lettere e rapporti di routine con i dati ricavati da un archivio creato esclusivamente con i moduli "Gestione dati personali (cod. DTXGX01)" o "Dati Statistici".

Configurazione raccomandata: A o C
D + E

A - MAZE - ING
Cod. DTXMX03 - L. 25.000

Se vi piacciono i labirinti, ecco il gioco per voi.
Opzioni diverse vi offrono una grande varietà di situazioni. Da semplici dedali allo caccio al topo. Tredici opzioni con 5.200 variazioni possibili.

Configurazione raccomandata: B

Oldies but Goodies (Vecchi ma buoni)
- Gioco II
Cod. DTXMX19 - L. 14.000

Una serie di giochi, che include Hamurabi, Hidden Paris, Peg Jump, Cerchi e croci tridimensionali e Word Sofori.

Configurazione richiesta: A

Munch Man
Cod. DTXMX16 - L. 45.000

Manovrate il Munch Man ottroverso un dedalo e cercate di raggiungere uno stimolatore prima di venire divorati dai quattro Haonos che stanno incalzando il Munch Man. Segnate punti callegando i passaggi con una catena continua oppure catturando gli ostuti Hoonos mentre il Munch Mon si ricarica con 10 stimolatori.

Configurazione raccomandata: B

Personal Record Keeping (Gestione dati personali)
Cod. DTXGX01 - L. 68.000

Permette di creare, mantenere ed utilizzare un sistema d'archivio computerizzato, utile e comodo per molte applicazioni, fra cui inventaria domestico, scadenze manutenzione autovettura, cartelle cliniche, oltre a un mezzo di consultazione completa per compleanni, onomastici, anniversari e altre date importanti.

Configurazione raccomandata: A o C D + E

Market Simulation (Simulazione di mercato)
Cod. DTXIX07 - L. 14.000

Due giocatori si trovano in concorrenza d'affari. Scegliete quanto pubblicità fare, quanti pezzi volete produrre, ecc. ed ottendete i risultati.
I mutamenti economici e sociali rendono estremamente realistico questo gioco istruttivo.

Configurazione richiesta: A

The Attack (Attacco)
Cod. DTXMX25 - L. 25.000

Vi attende il ruolo di capitano di un'astronave in una regione spaziale infestata da "spore" ed "estroterrestri" che vanno distrutti.
Manovrate l'astronave per evitare gli extraterrestri e lanciare missili per orientare il nemico.

Configurazione raccomandata: B

Configurazione:
A - Registratore a cassetto e cavetto di collegamento
B - Telecomandi o filo (coppia)
C - Memoria a dischi comprendente una scheda comando dischi e un comando dischi
D - Scheda interfaccia parallela o seriale RS232
E - Stampante a matrice (o altro tipo di stampante)

Connect Four (Filetto)
Cod. DTXMX08 - L. 25.000

Un impegnativo gioco di strategia verticale. I giocatori devono riuscire a collocare quattro contrassegni consecutivi in senso verticale, orizzontale o diagonale.

Zero Zap
Cod. DTXMX34 - L. 18.000

Flipper computerizzato dal ritmo rapido, con effetti sonori e di illuminazione. Potete creare voi stessi il campo di gioco.

Yahtzee
Cod. DTXMX33 - L. 25.000

Emozionante gioco di dadi che allo stesso tempo obbliga elementi di fortuna. I giocatori accumulano punti se escono certe combinazioni.

e inoltre 5 libri in lingua originale, estremamente utili per il tuo TI 99/4A:

Editore ARC soft - autore L. Turner:

36 Texas Instruments TI99/4A Programs for Home, School & Office
Cod. BASC002 - L. 20.000

101 Programming Tips & Tricks for the Texas Instruments TI 99/4A Home computer.
Cod. BASC001 - L. 20.000

Texas Instruments Home Computer Games Programs
Cod. BASC004 - L. 20.000

Texas Instruments Home Computer Graphics Programs
Cod. BASC003 - L. 22.000

Editore GRANADA - autore G. Morsholl:
Get more from the TI99/4A
Cod. BGRC001 - L. 15.000

Ritaglia ed invia a J. soft il tagliando sotto riportato, debitamente compilato in ogni sua parte.

Spett. J. soft - Via Rosellini, 12 - 20124 Milano (MI)
Tel. 02/6888228-6880841-6880842-6880843-683797

Ordino il seguente software/libri per il TI-99/4A:

cod.cod.cod.
cod.cod.cod.
cod.cod.cod.

per un totale di L. + L. 2.500 per contributo fisso di spese di spedizione.

Scelgo la seguente modalità di pagamento:

- ☐ pagherò in contrassegno al postino
- ☐ allego assegno (o contanti)
- ☐ verso l'importo sul C.C.P. n. 19445204
intestato a J. soft (allego ricevuta)

Nome

Cognome

Via n.

CAP Città Prov.

IN MILANO - VIA MASCHERONI, 14

IL VOSTRO "NUOVISSIMO" COMPUTER SHOP

Libri e riviste di elettronica e informatica.

CORSI CONTINUI TUTTO L'ANNO CON I MIGLIORI SOFTTERISTI



IL TELEFONO È 02-437.385